

豊四季タイニーBASIC

for MW25616L 実験用表示モジュール V0.03

リファレンスマニュアル RV01



作成日 2018年2月15日

作成者 たま吉さん

目次

豊四季タイニーBASIC for MW25616L 実験用表示モジュール V0.03 リファレンスマニュアル RV01.....	i
目次.....	i
1. 仕様や構成など.....	1
1.1 はじめに.....	1
1.2 関連情報.....	1
1.3 著作権・利用条件について.....	2
1.4 システム構成.....	2
1.5 仕様.....	3
1.6 ボードピン構成.....	4
1.7 メモリーマップ.....	5
1.8 VFD ディスプレイ (MW25616L) 対応.....	6
1.9 16 ドット日本語フォント (GT20L16J1Y) 対応.....	6
1.10 編集操作操作キー.....	7
1.11 キー入力コード.....	7
2. 利用環境設定.....	10
2.1 ファームウェアの書き込み.....	10
2.2 ターミナルソフトの通信条件の設定.....	11
2.3 接続とターミナルソフトの表示設定.....	12
2.4 コンソール画面の表示設定.....	12
2.5 プログラムの保存と読み込み.....	13
2.6 プログラム保存領域の初期化.....	13
2.7 パソコンからのプログラム読込.....	14
2.8 プログラムをパソコンに保存する.....	15
2.9 プログラムの自動起動方法.....	15
3. 使ってみよう！.....	16
3.1 入力可能状態は OK■ (カーソル).....	16
3.2 Hello,world を表示してみる.....	16
3.3 Hello,world を Hello,Tiny BASIC に変更してみる.....	16
3.4 PRINT の前に 10 を追加してプログラムにしよう.....	17
3.5 プログラムを実行してみよう.....	17
3.6 プログラムに追加する.....	17
3.7 行番号を並びなおす.....	18
3.8 やっぱり 10 行は不要、削除したい.....	18
3.9 プログラムを保存したい.....	19
3.10 保存されたプログラムを読み込み.....	19

4. プログラム構成要素の説明（コマンド・関数等）	20
4.1 プログラムの構造と形式	20
4.2 行と命令文	21
4.3 制御文	22
4.4 コマンド・関数	22
4.5 コマンド・関数一覧	23
4.6 数値	26
4.7 文字・文字列	26
4.8 文字列利用における制約	26
4.9 変数・配列変数	27
4.10 演算子	28
4.11 式	29
4.12 定数	29
5. 各制御文の詳細	32
5.1 GOTO 指定行へのジャンプ（制御命令）	32
5.2 GOSUB サブルーチンの呼び出し（制御命令）	33
5.3 RETURN GOSUB 呼び出し元への復帰（制御命令）	34
5.4 IF 条件判定（制御命令）	35
5.5 FOR TO ～ NEXT 繰り返し実行（制御命令）	36
5.6 END プログラムの終了（制御命令）	37
6. 各コマンド・関数の詳細	38
6.1 RUN プログラムの実行（システムコマンド）	38
6.2 RENUM 行番号の振り直し（システムコマンド）	39
6.3 LIST プログラムリストの表示	40
6.4 NEW プログラムの消去	41
6.5 LOAD 内部 EEPROM メモリからプログラムを読み（システムコマンド）	42
6.6 SAVE 内部 EEPROM メモリへのプログラム保存	43
6.7 FILES 内部フ EEPROM メモリ内保存プログラムの一覧表示	44
6.8 ERASE 内部フラッシュメモリ上のプログラム削除	45
6.9 EDIT スクリーンエディタ利用設定（システムコマンド）	46
6.10 FREE プログラム領域の残りバイト数の取得（数値関数）	48
6.11 REM コメント	49
6.12 LET 変数に値を代入	50
6.13 ABS 絶対値の取得（数値関数）	51
6.14 ASC 文字から文字コードへの変換（数値関数）	52
6.15 WASC 全角を含む文字から文字コードへの変換（数値関数）	54
6.16 LEN 文字列の長さの取得（数値関数）	56
6.17 WLEN 全角を含む文字列の長さの取得（数値関数）	57

6.18	CHR\$ 文字コードから文字への変換（文字列関数）	58
6.19	WCHR\$ シフト JIS コードから文字への変換（文字列関数）	59
6.20	BIN\$ 数値から 2 進数文字列への変換（文字列関数）	60
6.21	HEX\$ 数値から 16 進数文字列への変換（文字列関数）	61
6.22	STR\$ 変数が参照する文字列の取得・文字列の切り出し	62
6.23	WSTR\$ 変数が参照する全角を含む文字列の取得・文字列の切り出し	63
6.24	RND 乱数の発生（数値関数）	64
6.25	WAIT 時間待ち	65
6.26	TICK 経過時間取得（数値関数）	66
6.27	POKE 指定アドレスへのデータ書き込み	67
6.28	PEEK 指定アドレスの値参照（数値関数）	68
6.29	PRINT 画面への文字表示	69
6.30	INPUT 数値の入力	71
6.31	INKEY キー入力の読み取り（数値関数）	72
6.32	CLS 画面表示内容の全消去	73
6.33	COLOR 文字色の設定	74
6.34	ATTR 文字表示属性の設定	75
6.35	LOCATE カーソルの移動	76
6.36	VCLS VFD 表示内容の全消去	77
6.37	VDISPLAY VFD 表示のオン・オフ	78
6.38	VBRIGHT VFD 輝度設定	79
6.39	VMSG VFD にメッセージ文の表示	80
6.40	VSCROLL VFD に指定長空白の表示	81
6.41	VPUT VFD に直接データ送信	82
6.42	TONE 単音出力	83
6.43	NOTONE 単音出力停止	84
6.44	PLAY 音楽演奏(MML 文)	85
6.45	GPIO GPIO 機能設定	87
6.46	OUT デジタル出力	88
6.47	POUT PWM パルス出力	89
6.48	IN デジタル入力（数値関数）	90
6.49	ANA アナログ入力（数値関数）	91
6.50	I2CR I2C スレーブデバイスからのデータ受信（数値関数）	92
6.51	I2CW I2C スレーブデバイスへのデータ送信（数値関数）	94
7.	応用プログラム	97
7.1	RTC モジュール（DS3231 搭載）を使った時刻の表示	97

1. 仕様や構成など

1.1 はじめに

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」は整数型 BASIC インタープリタ実行環境です。Tetsuya Suzuki 氏が開発・公開している「TOYOSHIMI Tiny BASIC for Arduino」（以降オリジナル版と呼称）をベースに、MW25616L 実験用表示モジュール向けに移植し、機能拡張を行いました。「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」版のプログラムソースはオリジナル版と同様に公開しており、自由に改造等を行うことができます。

オリジナル版からの機能拡張としては、MW25616L 表示制御機能の他に、プログラム編集機能の強化（スクリーンエディット機能、プログラム保存機能、電子工作のための制御機能（デジタル入出力、PWM 出力、アナログ入力、I2C バス通信、シリアル通信）を追加しました。

本リファレンスマニュアルの構成は、前半に概要、システム構成、仕様、レガシーBASIC 環境を始めて利用する方向けの解説、中盤にリファレンス、後半に応用の内容となります。オリジナル版のタイニーBASIC の文法を含む、新しく追加した機能、コマンドについて解説・説明します。

1.2 関連情報

「豊四季タイニーBASIC（オリジナル版）」に関する情報

□ 開発者公開ホームページ

電腦伝説 Vintagechips 豊四季タイニーBASIC 確定版

<https://vintagechips.wordpress.com/2015/12/06/豊四季タイニーbasic 確定版/>

□ 開発者公開リソース

https://github.com/vintagechips/ttbasic_arduino/

□ 開発者執筆書籍

タイニーBASIC を C で書く

<http://www.socym.co.jp/book/1020>



原本のご紹介

タイニー BASIC を C で書く
パソコンでもマイコンでも走り、機能拡張し放題。

著者一鈴未哲哉

定価—3,200円+税

判型/ページ数—B5 変形/288 ページ

ISBN978-4-8026-1020-9

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」に関する情報

□ 公開サイト

豊四季タイニーBASIC for MW25616L 実験用表示モジュール (Tamakichi/ttbasic_mw25616l)

https://github.com/Tamakichi/ttbasic_mw25616l

□ MW25616L 実験用表示モジュールに関する情報

・「VFD ふぁん」さんの HP

<http://vfd-fun.blogspot.jp/>

1.3 著作権・利用条件について

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」の著作権は移植・機能拡張者の「たま吉」および、オリジナル版開発者の「Tetsuya Suzuki 氏」にあります。本著作物の利用条件は、オリジナル版「豊四季タイニーBASIC」に従うものとします。また、プログラムの一部に「VFD ふぁん」さんが公開しているプログラムソースを利用しています。

オリジナル版は GPL (General Public License) でライセンスされた著作物であり、「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」もそれに GPL ライセンスされた著作物とします。本著作物もこれに従うものとします。

また、下記の利用条件も守って下さい。

- 1) 2 次再配布においてはバイナリーファイルのみの配布を禁止、必ずプロジェクト一式で配布すること。
- 2) 上記配布において「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」の明記、配布元の URL を明記する。
- 3) 営利目的の NAVER 等のまとめサイト(まとめ者に利益分配のあるもの)へのいかなる引用の禁止

補足事項

オリジナル版において開発者 Tetsuya Suzuki 氏は次の利用条件を提示しています。

- 1) TOYOSHIMI Tiny BASIC for Arduino

https://github.com/vintagechips/ttbasic_arduino

の記載内容/

(C)2012 Tetsuya Suzuki

GNU General Public License

- 2) 豊四季タイニーBASIC のリファレンスを公開

<https://vintagechips.wordpress.com/2012/06/14/豊四季タイニーbasic リファレンス公開/>

の記載内容/


豊四季タイニーBASIC がもたらすいかなる結果にも責任を負いません。

著作権者の同意なしに経済的な利益を得てはいけません。

この条件のもとで、利用、複写、改編、再配布を認めます。

1.4 システム構成

「豊四季 Tiny BASIC for MW25616L 実験用表示モジュール」は、次の構成で利用可能です。



必要なもの

- パーソナルコンピューター：Windows 10 パソコン ※USB ポートが利用出来ること
- MW25616L 実験用表示モジュール本体
- Micro USB ケーブル
- シリアルコンソール対応ターミナルソフト：TeraTerm (V4.89 以降)

豊四季 Tiny BASIC インタープリター言語を利用したプログラム作成は、パソコン上のターミナルソフト上にてスクリーンエディタとコマンドを使って対話形式で行います。

シフト JIS を使った日本語文字の利用が可能です。ただし、Arduino (AVR) のメモリ容量の制約により、スクリーンエディタの画面サイズは横 48 x 縦 8 文字となります。

```
COM9:115200baud - Tera Term VT
ファイル(F) 編集(E) 設定(S) コントロール(O) ウィンドウ(W) ヘルプ(H)
40 @(1)='豊四季タイニーBASIC'
50 @(2)='VFD対応バージョン'
60 @(3)='只今開発中!'
70 @(4)='ちょっとメモリー的に厳しい..'
80 VBRIGHT 255
90 FOR S=0 TO 4
100 VCLS:VMSG 10,STR$(@(S))
110 FOR I=0 TO 255 STEP 15
```

Windows 以外の OS 環境でも同様の環境が構築できれば、利用出来ると思います。

1.5 仕様

(1) ソフトウェア仕様

項目		概要
プログラミング環境		ターミナル上での CUI (キャラクユーザーインタフェース) USB ケーブル経由シリアル接続のスクリーンエディタ+実行環境
ターミナルスクリーン画面		48×8 文字
B A S I C 言 語 仕 様	形式	整数型 BASIC (符号付 16 ビット整数 -32768~32767)
	プログラム領域	512 バイト
	利用文字コード	1 バイト ASCII コード (半角記号・英数字、半角カタカナ) 2 バイト文字 シフト JIS 表示可能文字はターミナル環境のフォント設定に依存
	利用可能変数	変数名 : 26 個 A~Z の英字 1 文字 26 個, 配列変数:16 個 @(0) ~ @(15)
	ユーザー開放 作業用メモリ	SRAM 64 バイト (先頭アドレス定数 MEM~) ※コマンド実行のワークエリアと共用のため内容の保持は出来ません。
	プログラム保存方式	内部 EEPROM メモリ 2 本 (ページ 0, 1 指定)
	フォントデータ	フォント ROMGT30L16J1Y 日本語 JIS 第 1、第 2 フォント ※MW25616L (VFD) 表示専用

(2) ハードウェア仕様

項目	概要
動作クロック	16MHz
VFD	MW25616L 256x16 ドット モノラル表示、輝度調節 0~255 段階調節可能
加速度センサー	MMA8653 (3方向)
USB I/F	USB-シリアル通信
シリアル通信	1 チャンネル 115200、データ長 8 ビット、 ストップビット 1 パリティ None
I2C インタフェース	1 チャンネル (マスター利用のみ) アドレス 7 ビット、400kbps
GPIO	<ul style="list-style-type: none"> デジタル入力 1 ビット× 17 ポート デジタル出力 1 ビット× 17 ポート PWM 出力 3 チャンネル (周波数固定、デューティ比 1/255 刻み)
アナログ入力	5 チャンネル (分解能 10 ビット)
単音出力	デジタル矩形波出力による簡易単音再生

(3) サポートする周辺機器

- 圧電スピーカー : TONE、NOTONE、PLAY コマンドによる単音出力をサポートします。
- シリアルポート : USB-シリアル変換モジュールを使ったシリアル通信をサポートします。

1.6 ボードピン構成

MW25616L 実験用表示モジュールにて解放されている端子の利用割当は以下の通りです。

Pin 0	: RxD 専用
Pin 1	: TxD 専用
Pin 2	: デジタル入出力(2)
Pin 5	: デジタル入出力(5)、PWM(5)
Pin 6	: デジタル入出力(6)、PWM(6)
Pin 7	: デジタル入出力(7)、保存プログラム自動起動判定用 (HIGH 自動起動、LOW 自動起動は行わない)
Pin 8	: デジタル入出力(8)、単音出力用 (圧電スピーカー接続)
A3	: デジタル入出力(17)、アナログ入力(17)
A4	: デジタル入出力(18)、アナログ入力(18)、I2C(SDA)
A5	: デジタル入出力(19)、アナログ入力(19)、I2C(CLK)
A6	: デジタル入出力(20)、アナログ入力(20)
A7	: デジタル入出力(21)、アナログ入力(21)
REST	: RESET 入力
SCK	: 使用不可
MISO	: 使用不可
MOSI	: 使用不可

1.7 メモリーマップ

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」ではフラッシュメモリ領域 32k バイト、SRAM は 2k バイトの領域を利用しています。ここでは、これらの領域の内訳を示します。

フラッシュメモリ

32k バイト全領域を Arduino ブートローダーと豊四季タイニーBASIC にて占有しています。

SRAM

2k バイト全領域を豊四季タイニーBASIC にて占有しています。

仮想アドレス

タイニーBASIC では、仮想アドレス利用することで SRAM 上のデータの参照・書き込みを行うことができます。

仮想アドレスは次の構成となります。

仮想アドレス	定数名	領域サイズ(バイト)	用途
\$0000	VRAM	384	キャラクタスクリーン表示用メモリ 横 48×縦 8 = 384 バイト
\$1900	VAR	52	変数領域 (A~Z) 2×26 = 52 バイト
\$1AA0	ARRAY	32	配列変数領域 (@ (0) ~ @ (15))
\$1BA0	PRG	512	プログラム領域
\$2BA0	MEM	64	ユーザーワーク領域

MEM (ユーザーワーク領域) は、下記のコマンドのワーク領域としても利用しています。

FILES、VMSG

プログラム中でユーザーワーク領域を利用している場合、上記のコマンドを実行するとデータが破壊されることに注意して利用して下さい。

各領域のアドレス先頭は定数 VRAM, VAR, ARRAY, PRG, MEM に定義されています。

プログラムから上記の定数を利用して利用することが出来ます。

1.8 VFD ディスプレイ (MW25616L) 対応

ボード搭載 VFD ディスプレイ MW25616L を制御するためのコマンドをサポートします。

コマンド一覧

- VBRIGHT : 輝度を 0(0%)~255(100%) の範囲で設定します。
例: VBRIGHT 100
- VDISPLAY : 表示のオン・オフを設定します。
例: VDISPLAY OFF
- VMSG : 指定したメッセージ文を指定速度でスクロール表示します。
例: VMSG 10, "日本語のメッセージ表示テスト"
- VSCROLL : 指定長の空白を指定速度でスクロール表示します。
例: VSCROLL 256, 10
- VPUT : 表示バイトデータを指定速度で直接指定します。
例: VPUT MEM, 16, 10

各コマンドの詳細については、「6 各コマンド・関数の詳細」を参照して下さい。

1.9 16 ドット日本語フォント (GT20L16J1Y) 対応

VFD ディスプレイ MW25616L 表示専用に日本フォントを搭載しています。

VMSG コマンドにて指定したメッセージ文に対応するフォントデータを随時参照利用します。

1.10 編集操作操作キー

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」では、シリアルコンソール画面でのフルスクリーン編集をサポートしています。

編集キー一覧

編集キー	機能
[ESC]	プログラム中断、シリアルコンソールでは要2回押し
[F1]	画面の全消去
[F2]	カーソル位置の行消去
[F3]	カーソルの次行に空行挿入
[F3]	プログラムの中断
[F5]	画面の再表示
[F7]	カーソル以降の文字列を次の行に分割する
[F8]	下の行の文字列を結合する
[BackSpace]	カーソル前の文字の削除
[Insert]	上書きモード、挿入モードも切り替え
[Home]	カーソルを行の先頭に移動
[END]	カーソルを行の末尾に移動
[PageUP]	カーソルを画面右上に移動、画面のスクローダウン
[PageDown]	カーソルを表示している最終行に移動、スクロールアップ
[Del]	カーソル位置の文字の削除
[←]	カーソルを左に移動
[→]	カーソルを右に移動
[↑]	カーソルを上移動
[↓]	カーソルを下移動
[Enter]	行入力の確定、改行
[NumLock]	テンキーのロック、ロック解除
[カタカナ/ひらがな/ローマ字]	カタカナ入力のON、OFF
[Ctrl] + C	プログラム中断
[Ctrl] + D	カーソル位置の行削除
[Ctrl] + K	カタカナ入力のON、OFF
[Ctrl] + L	画面の全消去
[Ctrl] + N	カーソルの次行に空白挿入
[Ctrl] + R	画面の再表示
[Ctrl] + X	カーソル位置の文字の削除


1.11 キー入力コード

INKEY()関数等で入力したキーを判定する際は、キー入力コードを用います。

大部分のキー入力コードは入力する文字コード、ASCIIコードと一致します。カーソルキー等の入力制御用のキーは、「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」固有のコードとなります。

□ キー入力コードが取得出来ないキー

赤枠のキーは制御用または利用不可キーのためキーコードの取得が出来ません。



106/109 キーボード (※画像は ウィキペディア <https://ja.wikipedia.org/wiki/キー配列> より拝借)

□ キー入力コード一覧

キーボード入力にて取得出来るキー入力コードを以下に示します。

コードに数値の記載のあるキーに限りコードの取得が可能です。

IME を使った全角入力もサポートしています。

キー	併用なし		Shift		CTRL	
	文字	コード	文字	コード	文字	コード
[Backspace]		8				
[Enter]		13				
[Tab]		9				
[CapsLock]						
[Shift]						
[Ctrl]						
[Win]						
[Alt]						
[無変換]						
[空白 (Space)]	空白	32				
[変換]						
[かなひらがなローマ字]						
[メニュー]						
[Insert]		5				
[Home]		1				
[PageUp]		17				
[Delete]		2				
[End]		15				
[PageDown]		16				
[↑]		28				
[↓]		29				
[←]		31				
[→]		30				
[PrintScreen]						
[ScrollLock]						
[Pause/Break]						
[! め]	1	49	!	33		
[" ふ]	2	50	"	34		
[# ああ]	3	51	#	35		
[\$ うう]	4	52	\$	36		
[% ええ]	5	53	%	37		
[& おお]	6	54	&	38		
[' 7 やや]	7	55	'	39		
[(8 ゆゆ]	8	56	(40		
[) 9 よよ]	9	57)	41		
[0 をわ]	0	48				
[- ほ]	-	45	=	61	CTRL -	29
[~ へ]	^	94	~	126		
[¥]	¥	92		124		

はサポートしていないキーです。

キー	併用なし		Shift		CTRL	
	文字	コード	文字	コード	文字	コード
[Q た]	q	113	Q	81	CTRL Q	17
[W て]	w	119	W	87	CTRL W	23
[E いい]	e	101	E	69	CTRL E	5
[R す]	r	114	R	82	CTRL R	18
[T か]	t	116	T	84	CTRL T	20
[Y ん]	y	121	Y	89	CTRL Y	25
[U な]	u	117	U	85	CTRL U	21
[I に]	i	105	I	73	CTRL I	9
[O ら]	o	111	O	79	CTRL O	15
[P せ]	p	112	P	80	CTRL P	16
[`@`]	@	64	`	96		
[{ [`°]	[91	{	123		
[A ち]	a	97	A	65	CTRL A	1
[S と]	s	115	S	83	CTRL S	19
[D し]	d	100	D	68	CTRL D	4
[F は]	f	102	F	70	CTRL F	6
[G き]	g	103	G	71	CTRL G	7
[H く]	h	104	H	72	CHRL H	8
[J ま]	j	107	J	74	CTRK J	10
[K の]	k	107	K	75	CTRK K	
[L り]	l	108	L	76	CTRK L	12
[+;れ]	;	59	+	43		
[*:け]	:	58	*	42	CTRL *	28
[]}] む]]	93	}	125		
[Z っつ]	z	122	Z	90	CTRL Z	26
[X さ]	x	120	X	88	CTRL X	24
[C そ]	c	99	C	67	CTRL C	
[V ひ]	v	118	V	86	CTRL V	22
[B こ]	b	98	B	66	CTRL B	2
[N み]	n	110	N	78	CTRL N	14
[M も]	m	109	M	77	CTRL M	13
[<、ね]	,	44	<	60		
[>。る]	.	46	>	62	CTRL >	30
[?/・め]	/	47	?	63	CTRL ?	31
[¥_ろ]	¥	92	_	95		

2. 利用環境設定

本章では、「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」を使ってプログラムの開発を行うための環境及びその設定について、解説します。

2.1 ファームウェアの書き込み

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」を利用するには、Arduino IDE にて配布スケッチをコンパイルして書き込む必要があります。

必要環境

- ① Arduino IDE が利用可能なパソコン
- ② Arduino IDE 1.8.5 以降
- ③ MW25616L 実験用表示モジュール

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」用ファームウェア書き込み手順

- ④ 「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」プロジェクトを下記のリンクからファイル **ttbasic_mw25616l-master.zip** をダウンロードします。
https://github.com/Tamakichi/ttbasic_mw25616l/archive/master.zip
- ⑤ ダウンロードしたファイルを解凍します。
エクスプローラーにて **ttbasic_mw25616l-master.zip** を選択し、
マウス右クリックにて開いたメニューから「すべて展開」を選択することで解凍することが出来ます。
解凍したフォルダ内の **ttbasic** フォルダがスケッチとなります。
- ⑥ Arduino IDE を起動し、スケッチ **ttbasic** を読み込みます。
- ⑦ MW25616L 実験用表示モジュールを接続します。
- ⑧ Arduino IDE でスケッチをコンパイルし、書き込みます。

以上の操作で書き込みが完了します。

次のステップとしては、ターミナルソフトを使って MW25616L 実験用表示モジュールに接続します。

2.2 ターミナルソフトの通信条件の設定


ここでは、Windows 10 パソコン上でターミナルソフト TeraTerm を利用する場合の設定について解説します。
TeraTerm をお使いの場合は、下記のサイトからダウンロード・インストールして下さい。

通信条件の設定

ファームウェアを書き込んだ後、MW25616L 実験用表示モジュールを USB ケーブルにて続した状態にします。

①通信条件の設定

ポート	設定値
ポート	各自の環境に合わせて設定
ボー・レート	115200(任意)
データ長	8 bit
パリティビット	無し
ストップビット	1bit
フロー制御	無し



USB 接続の場合、ボー・レートの設定任意です。


ボー・レートの設定は無視され USB 通信の最適な速度で通信を行うようです（推測）。

TeraTerm の設定は、メニュー[設定] - [シリアルポート]から[シリアルポート設定]画面を開いて行います。

②言語等に関する設定

「豊四季タイニー BASIC for MW25616L 実験用表示モジュール」には 1 バイトコードの半角カタカナの利用をサポートします。半角カタカナ利用のためには、利用する文字列コードとしてシフト JIS を指定します。


[設定] - [全般設定] - [全般設定]画面 の設定



言語 : Japanese

言語 UI : Japanese.lng

[設定] - [端末の設定] - [端末の設定]画面 の設定



端末サイズ : 48x8 以上に設定

漢字-受信 : SJIS

漢字-送信 : SJIS

改行コード受信 : AUTO、送信 CR

ウィンドウサイズ : ☒チェックを入れる

2.3 接続とターミナルソフトの表示設定

ターミナルソフトの設定を行ったら、MW25616L 実験用表示モジュールがリセットされ、次のターミナル上に起動メッセージが表示されます。

```
TOYOSHIKI TINY BASIC
ARDUINO Extended version 0.02
OK
```


表示されない場合は、ケーブル及びターミナルソフトの通信条件の設定を見直して下さい。

ターミナルソフトのウィンドウサイズをマウス操作等で変更すると表示内容が消えてしまいます。

この場合は[F5]キーを押して、再表示して下さい。

表示された文字が小さい場合はフォントを調整して下さい。

フォントは、文字セットに日本語が設定可能なフォントを利用して下さい。



2.4 コンソール画面の表示設定

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」のコマンドにて、スクリーンエディタ機能の有効・無効設定、フォントの色、背景色の設定を行うことが出来ます。

コマンド

コマンド	説明
EDIT モード	スクリーンエディタ機能の設定を行います。 モード：ON エディタ有効（デフォルト）、OFF エディタ無効
COLOR n [, m]	テキスト文字色、背景色を設定します。 n：文字色 0～8 m：背景色 0～8 COLOR コマンドの色指定については、「6.33 COLOR 文字色の設定」を参照して下さい。 COLOR コマンド実行後、CLS コマンドを実行するか、[F1]キーを押すことで画面全体に色設定が反映されます。
ATTR n	文字表示属性を設定します。 n: 0 ノーマル、 1 下線、 2 反転、 3、ブリンク、 4 ボールド
CLS	画面表示のクリア キーボードの [F1]キー、CTRL+L キーでもクリアできます。

(補足) ターミナルソフト TeraTerm の設定により、利用するフォントの種類・サイズ等の指定を行うことが出来ます。
TeraTerm の設定については、TeraTerm のマニュアルを参照して下さい。

2.5 プログラムの保存と読み込み

作成したプログラムは内部 EEPROM メモリに保存可能です。

内部 EEPROM メモリへの保存 [SAVE](#) プログラム番号(0~1)

内部 EEPROM メモリからの読み込み [LOAD](#) プログラム番号(0~1)

内部 EEPROM メモリに保存しているプログラムのリスト表示： [FILES](#)

```
files
0:CLS
1:(none)
OK
```

FILES コマンドはプログラムの先頭行を表示します。

プログラム番号の格納領域にプログラムが保存されていない場合は、(none)が表示されます。

内部 EEPROM メモリ上のプログラムは ESASE コマンドにて削除することが出来ます。

マイコンボードの リセット時に予め指定したプログラムを起動することも可能です。

詳細は、「2.9 プログラムの自動起動」を参照して下さい。

2.6 プログラム保存領域の初期化


内部 EEPROM メモリの初期化は、[ERASE](#) コマンドを利用します。

全領域を初期化するには次のコマンドを実行します。

```
ERASE 0,1
OK
```

2.7 パソコンからのプログラム読み込み

パソコンからマイコンボードにプログラムを転送する方法としては、
ターミナルソフトの画面にプログラムソースコピー＆ペーストする方法が簡単でおすすめです。
この際、シリアルポートの設定にて[送信遅延]を下記のように設定して下さい。




試しに、次のプログラムをコピーしてパソコンから MW25616L 実験用表示モジュールにプログラムを転送してみましょう。

```

10 'VDF表示サンプル
20 TONE 440,100:TONE 880,100
30 @(0)="熱いぞ！埼玉！"
40 @(1)="豊四季タイニーBAS I C"
50 @(2)="VFD対応バージョン"
60 @(3)="只今開発中！"
70 @(4)="ちょっとメモリーの的に厳しい.."
80 VBRIGHT 255
90 FOR S=0 TO 4
100 VCLS:VMSG 10,STR$(@(S))
110 FOR I=0 TO 255 STEP 15
120 VBRIGHT 255-I:WAIT 40
130 NEXT I
140 WAIT 500
150 VBRIGHT 255:WAIT 500
160 VSCROLL 256,3
170 WAIT 500
180 NEXT S
190 GOTO 80

```

TeraTerm では、メニュー [編集] - [貼り付け<CR>] か ALT+R キーにて貼り付けることができます。



(注意) プログラムの読み込みが終わったら、送信遅延の設定は元に戻して下さい。プログラム実行時にコンソール画面からの文字入力を行うコマンド INPUT、INKEY()にて遅延が発生し、意図しない動作をする場合があります。

2.8 プログラムをパソコンに保存する

SRAM 上、内部 EEPROM メモリ上のプログラムをパソコンに保存するには、ターミナル上で LIST コマンドを実行し、ターミナルソフトをスクロールする等して、テキストを選択してコピー＆ペースト等を行ってメモ帳等に張り付けて保存して下さい。

プログラム行数が長い場合は、コマンド EDIT OFF を実行して、てスクリーンエディタ機能を無効にすることで、より多くの行の表示を行うことが出来ます。

```

COM9:115200baud - Tera Term VT
ファイル(F) 編集(E) 設定(S) コントロール(O) ウィンドウ(W) ヘルプ(H)
OK
>list
1 'VFDジグザグ表示
10 VCLS
20 @(0)=1
30 FOR I=0 TO 14
40 VPUT ARRAY,2,5
50 @(0)=@(0)<<1
60 NEXT I
70 FOR I=0 TO 14
80 @(0)=@(0)>>1
90 VPUT ARRAY,2,5
100 NEXT I
110 GOTO 30
OK
>

```

ファームウェアの更新を行う場合は事前に、内部 EEPROM メモリに保存されているプログラムをパソコンに保存して下さい。ファームウェア更新後は、保存しているプログラムが動かなくなる場合があります。

2.9 プログラムの自動起動方法

MW25616L 実験用表示モジュールの起動時またはリセット時にフラッシュメモリー上のプログラム番号 No.0 のプログラムを起動することが出来ます。その方法について説明します。

方法（自動起動条件）

電源 ON またはリセット時に No7 ピン が HIGH(5V)の場合、プログラム番号 No.0 を自動起動します。

このピンはスケッチの修正により、別のピンに変更することが可能です

3. 使ってみよう！

この章では、「レガシーBASIC（古き良き時代の BASIC 言語）の雰囲気を知らない方々にその雰囲気をつかんでもらいましょう」ということで、軽いフットワークで解説進めます。

準備は OK？ 前章の実行環境の設定は完璧？ それでは電源を入れて使ってみよう！

3.1 入力可能状態は OK■（カーソル）

```
TOYOSHIKI TINY BASIC
ARDUINO Extended version 0.02
OK
■
```

この状態は、「いつでもこいや〜」状態。
きみのコマンド入力を待っている状態。
何か打ち込んでみよう！

3.2 Hello,world を表示してみる

直接、次の文字を打ってみよう。

```
print "Hello,world" Enter
```

入力ミスの修正は[BS]キー、[DEL]キー、カーソルキーなどいつものキーが使えるよ。
入力出来たら、最後に[Enter]キーを押してみよう。

```
print "Hello,world" Enter
Hello,world
OK
■
```

表示出来たね。PRINT は文字列を表示するコマンドなのです！
入力は大文字、小文字どちらでも OK！

3.3 Hello,world を Hello,Tiny BASIC に変更してみる

カーソルキー[↑][↓][←][→]でカーソルを動かして、world の w に移動してみよう。

```
print "Hello,world"
Hello,world
```

[DEL]キーで world を削除して、代わりに Tiny BASIC と入力しよう。
入力後に、[Enter]キーを押すのを忘れずに！

```
print "Hello,Tiny BASIC" Enter
Hello,Tiny BASIC
OK
■
```

こんな感じで[Enter]が入力確定とコマンド実行の合図だ。
[Enter]を入力する際、カーソルは行のどこにあっても構わない。

3.4 PRINT の前に 10 を追加してプログラムにしよう

再びカーソル操作で print の前に 10 を挿入して[Enter]キーを押してみよう。

```
10print "Hello,Tiny BASIC" Enter
Hello,Tiny BASIC
OK
```

これでプログラムとして登録された。LIST コマンドで確認してみよう。
カーソルを OK の下の空き行に移動して、list と入力して[Enter]キーを入力

```
10 print "Hello,Tiny BASIC" Enter
Hello,Tiny BASIC
OK
list Enter
10 PRINT "Hello,Tiny BASIC"
OK
■
```

登録したプログラムの内容が表示されたね。

まだ、行番号 10 しか登録していないので、その 1 行だけの表示だ。

このように、先頭に数字(=行番号)をつけると行番号の後ろのコマンドがプログラムとして登録されるのだ。

3.5 プログラムを実行してみよう

う〜ん、気分的に、画面を綺麗にしてから表示したくなった。

画面をきれいにしてから、3回実行してみよう。

```
cls Enter
run Enter
Hello,Tiny BASIC
OK
run Enter
Hello,Tiny BASIC
OK
run Enter
Hello,Tiny BASIC
OK
```

3回同じ処理が実行出来たね。

CLS は画面を消すコマンド、RUN はプログラムを実行。よく使うコマンドだよ。

3.6 プログラムに追加する

今の“画面きれいにしてから、3回表示”をプログラムにやらせてみよう！

次のように入力して、プログラムとして登録するよ。

```
5 cls Enter
7 for i=1 to 3 Enter
20 next i Enter
```

入力したプログラムを確認！

```
List Enter
5 CLS
7 FOR I=1 TO 3
10 PRINT "Hello,Tiny BASIC"
20 NEXT I
OK
```

今入力した 5,7,10 行が追加されたね。プログラムは行番号順に表示されるよ。

それでは実行！ run コマンドで実行するよ。

```
run Enter
Hello,Tiny BASIC
Hello,Tiny BASIC
Hello,Tiny BASIC
OK
```

3 行表示出来た。

プログラムの実行順番は行番号順。CLS が最初に実行されるよ。

FOR 文は繰り返しを行う命令。3 回 10 行の表示を行ったよ。

3.7 行番号を並びなおす

もう一回、プログラムを確認しよう。

```
List Enter
5 CLS
7 FOR I=1 TO 3
10 PRINT "Hello,Tiny BASIC"
20 NEXT I
OK
```

行番号の 5,7,10,20 と並びがちょっと美しくないですね。

直してみよう。

RENUM コマンドを使ってプログラムの行番号を 10 ずつに整えるよ。

```
renum Enter
OK
List Enter
10 CLS
20 FOR I=1 TO 3
30 PRINT "Hello,Tiny BASIC"
40 NEXT I
OK
```

きれいに並んだね。

3.8 やっぱり 10 行は不要、削除したい

やっぱり、直前の表示内容は消したくない。

10 行はいらない。削除したい。行番号だけ入力して[Enter]を押してみよう。

```
10 Enter
```

これで 10 行が削除されるよ。確認してみよう。

```
List Enter
20 FOR I=1 TO 3
30 PRINT "Hello,Tiny BASIC"
40 NEXT I
OK
```

10 行が無くなったね。

こんな感じ行番号だけの入力で、行削除。よく使う操作だよ。

DELETE コマンドを使っても行の削除ができるよ。

```
DELETE 10 Enter
```

20 行から始まるのは美しくない。行番号をまた振り直そう。

```
renum Enter
OK
List Enter
10 FOR I=1 TO 3
20 PRINT "Hello,Tiny BASIC"
30 NEXT I
OK
```

3.9 プログラムを保存したい

今作成したプログラムは SRAM 上に保存されているのだ。

マイコンの電源を切ると消えてしまう・・・

消えないで欲しい・・・。そんな時、SAVE コマンドを使うよ。

SAVE コマンドはプログラムをフラッシュメモリに保存するよ。

フラッシュメモリに保存されたプログラムは電源を切っても消えないよ。

次のコマンドを打ってみましょう。

```
save Enter
OK
```

これで保存できたはずだ。

3.10 保存されたプログラムを読み込み

本当に保存されているか確認してみましょう。

SRAM 上のプログラムを消してフラッシュメモリを読み込みと試してみよう。

NEW コマンドを打ってプログラムを初期化するよ。

```
new Enter
OK
```

これで消えたはず。

LIST コマンドで確認。

```
list Enter
OK
```

何も出てこない。本当に消えた。

次に LOAD コマンドでフラッシュメモリからプログラム読み込むよ。

```
loadEnter
OK
list Enter
10 FOR I=1 TO 3
20 PRINT "Hello,Tiny BASIC"
30 NEXT I
OK
```

成功！ 復活出来た！・・・

保存しているプログラムは FILES コマンドで確認できるよ。

0 番に保存されているよ。

```
files Enter
0:FOR I=1 TO 3
1:(none)
OK
```

これで基本的な操作は終了だ。

基本はバッチリ、準備万端！ Tiny BASIC を使った更なるプログラミングをエンジョイしよう！

4. プログラム構成要素の説明（コマンド・関数等）

本章では、「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」のプログラム言語について解説します。

4.1 プログラムの構造と形式

次のプログラムリストは、VFDにメッセージを表示するプログラムです。

プログラムを実行すると0～5までの数字がスクロール表示されます。

```
1 '数字のスクロール表示
10 VCLS:VBRIGHT 255:VDISPLAY ON
20 FOR I=1 TO 5
30 VMSG 5,I;"だよ～"
40 VSCROLL 256,5
50 NEXT I
```

プログラムの構造的は・・・

プログラムは複数の行（行番号＋命令文）で構成されます。

例）1'数字のスクロール表示 ：プログラム内容を説明するコメントです。

例）30 VMSG 5,I;"だよ～" ： VFD に文字を表示します。

命令文は1つまたは複数のコマンドや式で構成されます。

例）30 IF !IN(BTNA) GOSUB "@ShowTime" ： A ボタンが押されたらサブルーチンを呼び出しています。

コマンドはコマンド名と引数で構成されます。

例）10 VCLS:VBRIGHT 255:VDISPLAY ON ： VFD の表示の初期化を行っています。

引数は式、変数、定数にて構成されます。

例）30 VMSG 5,I;"だよ～" ： 変数に格納された数値に文字列定数を結合して VFD に表示しています。

基本的に、プログラムは行番号順にシーケンシャルに実行されます。

例）1 行から 20 行は順番に実行されます。

制御命令により、分岐や繰り返しを行うことが出来ます。

例）50 行は NEXT I 文にて 20 行に戻ります。

このようにプログラムは様々な要素で構成れます。

次節からは、これらの各構成要素について解説します。

4.2 行と命令文

□ 行

行とは利用者がコンピュータに対して指示を行う単位です。行は命令文で構成されます。

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」では、行の単位で利用者からの指示を処理します。

スクリーンエディタ、およびコンソールから、

```
PRINT "Hello" Enter
Hello
OK
```

のように **Enter** キーの入力により、行の単位で指示を受け取りその処理を行います。

また、行の先頭に行番号を付けた場合、

```
10 PRINT "Hello" Enter
```

「指定した行番号で行（命令文）をプログラムに登録せよ」

との指示であると判断し、命令文は実行せずに、行（命令文）をプログラムとしてメモリーに登録します。

```
10 Enter
```

のように行番号だけを指定した場合は、

「指定した行番号で行（命令文）を空にせよ（削除せよ）」

との指示であると判断し、該当する行をプログラムから削除します。

□ 命令文

命令文とはコマンド、式、関数、コメント、ラベルを組み合わせて記述した命令です。

コマンドはコロン:を使って1行に複数記述することができます。

```
10 I=I+1:LOCATE 0,I:COLOR 7,0:PRINT "Hello":GOTO 50:'サンプル
```

コメント文

文において、REM、および省略形のシングルクォーテーション'を使った以降の文はコメントとなります。

例：

```
10 'Smple program
20 REM Sample program
30 PRINT "Hello":'print hello
```

ラベル

行の先頭のダブルクォーテーションで囲った文字列はラベルとして動作します。

ラベル自体は実行時に何も行いません。GOTO 文、GOSSUB 文のジャンプ先の指定で利用します。

ラベルの後ろにはコマンドを続けて記述することができます。

例：

```
10 "LOOP":PRINT "Hello"
20 GOTO "LOOP"
```

4.3 制御文

制御文は制御命令を使ったプログラムの記述文です。複数のキーワード、式の組み合わせで構成されます。プログラムは RUN コマンドを実行することにより、通常は先頭行から行番号順に逐次実行されます。この逐次実行は制御文を用いることで条件分岐、繰り返しを行うことが出来ます。

FOR 文の例：繰り返し処理を行う

```
10 FOR A=0 TO 10 STEP 5
20 PRINT A
30 NEXT A
```

以下に制御命令の一覧を示します。

制御命令	説明
GOTO 行番号 GOTO ラベル	指定行へのジャンプ
GOSUB 行番号 GOSUB ラベル	サブルーチンの呼び出し
RETURN	GOSUB 文サブルーチン呼び出しからの復帰
IF 条件式 実行文 1 IF 条件式 実行文 1 ELSE 実行文 2	条件判定
FOR 変数=初期値 TO 最終値 実行文 (複数行可能) NEXT 変数 FOR 変数=初期値 TO 最終値 STEP 増分 繰り返し実行文 NEXT 変数	繰り返し
END	プログラムを終了する

各制御命令の詳細については、「5 各制御文の詳細」を参照して下さい。

4.4 コマンド・関数

コマンド

コマンドとは何らかの処理を行うプログラム要素です。コマンドには一般コマンドとシステムコマンドの2種類があります。一般コマンドはコマンドラインからの直接利用とプログラム内での利用が可能です。

システムコマンドは、プログラムの管理を行うコマンドです。プログラム中で利用した場合、不整合が生じるため、コマンドラインでのみ利用可能としています。

システムコマンドをプログラム中で利用した場合、エラー("Illegal command")となります。

例：プログラム内にシステムコマンド [RENUM](#) を記述

```
10 ? "SAMPLE"
20 RENUM 10
30 END
RUN
Illegal command in 20
20 DELETE 10
OK
```

プログラム中にプログラムの行番号の付け替えを行うと不具合が生じるため、実行することは出来ません。

関数

関数とは何らかの値を返す命令文です。関数には数値関数と文字列関数の2種類があります。

数値関数は数値を返す関数です。文字列関数は文字列を返す関数です。

数値関数はコマンドの引数や式のなどの数値として利用します。

文字列関数は PRINT、VMSG の引数にて利用できます。

例：関数の利用

```
10 FOR I=ASC("A") TO ASC("F")
20 PRINT "code=";HEX$(I,2)
30 NEXT I
40 END
```

```
run
code=41
code=42
code=43
code=44
code=45
code=46
OK
```

上記のプログラムは文字“A”～“F”の文字コードを 16 進数へ表示するプログラムです。

10 行にて数値関数 [ASC\(\)](#) を利用しています。20 行で文字列関数 [HEX\\$\(\)](#) を利用しています。

関数単独での利用は出来ません。コマンドや関数の引数、式中でのみ利用可能です。

例：関数の利用は文法エラーとなる

```
ASC("A")
Syntax error
OK
```

4.5 コマンド・関数一覧

機能別に分類したコマンド及び関数の一覧を示します。

各コマンド及び関数の利用方法の詳細については、「7.各コマンド・関数の詳細」を参照して下さい。

□ システムコマンド

RUN	プログラムの実行
RENUM	行番号の振り直し
RENUM 先頭行番号	先頭行番号：振り直し開始先頭行
RENUM 先頭行番号,間隔	間隔：行間の増分

□ プログラム関連

コマンド

LIST	プログラムリストの表示
LIST 開始行番号	開始行番号：表示を開始する行
LIST 開始行番号,終了行番号	終了行番号：表示を終了する行
NEW	プログラムの消去
LOAD	内部フラッシュメモリからプログラム読み込み
LOAD プログラム番号	プログラム番号：読み出し元番号 (0～7)
SAVE	内部フラッシュメモリ/SD カードへのプログラム保存
SAVE [プログラム番号]	プログラム番号：保存先番号 (0～9)
FILES	内部フラッシュメモリのプログラム一覧表示
FILES 開始[,終了]	開始,終了:プログラム番号
REM [コメント文] or '[コメント文]	コメント '省略形の' (シングルクォーテーション)も可能
LET 変数=式	変数・配列変数に値を代入(LET は省略可能)
LET @(添え字)=n1[,n2,n3,n4…]	配列変数では連続代入指定が可能
ERASE プログラム番号	内部フラッシュメモリ上のプログラム削除
ERASE 開始プログラム番号,終了プログラム番号	
EDIT モード	スクリーンエディタ利用設定

数値関数

ABS (整数)	絶対値の取得
ASC (変数) ASC (変数,文字位置) ASC (文字列) ASC (文字列,文字位置)	変数が参照する文字列の取得/文字列の切り出し 例 A=ASC("ABCD",2) S="ABC";B=ASC(S,2)
WASC (変数) WASC (変数,文字位置) WASC (文字列) WASC (文字列,文字位置)	変数が参照する文字列の取得/文字列の切り出し (全角 SJIS 対応) 例 A=WASC("あ ABCD",2) S="あ ABC";B=ASC(S,2)
FREE ()	プログラム領域の残りバイト数の取得
INKEY ()	キー入力の読み取り
RND ()	乱数の発生
LEN (変数) LEN (文字列)	文字列長の取得
WLEN (変数) WLEN (文字列)	文字列長の取得 (全角 SJIS 対応)
GRADE (値, 配列 No, 個数)	指定した値の等級を求める

文字列関数

CHR\$ (文字コード[,文字コード,..., 文字コード])	文字コードから文字への変換
WCHR\$ (文字コード[,文字コード,..., 文字コード])	文字コードから文字への変換 (全角 SJIS 対応)
BIN\$ (数値[,桁指定])	数値から 2 進数文字列への変換
HEX\$ (数値[,桁指定])	数値から 16 進数文字列への変換
STR\$ (変数) STR\$ (変数,先頭,長さ) STR\$ (文字列) STR\$ (文字列,先頭,長さ)	変数が参照する文字列の取得
WSTR\$ (変数) WSTR\$ (変数,先頭,長さ) WSTR\$ (文字列) WSTR\$ (文字列,先頭,長さ)	変数が参照する文字列の取得 (全角 SJIS 対応)

□ 時間待ち・時間計測関連

コマンド

WAIT ミリ秒	時間待ち
--------------------------	------

数値関数

TICK ()	起動からの経過時間取得 (ミリ秒単位)
TICK (1)	起動からの経過時間取得 (秒単位)

□ 記憶領域操作関連

コマンド

POKE アドレス,データ[,データ,...データ]	指定アドレスへのデータ書き込み
--------------------------------------------	-----------------

数値関数

PEEK (アドレス)	指定アドレスの値参照
-----------------------------	------------

□ コンソール画面表示関連

コマンド

PRINT [#n,] 数値・文字列[: 数値・文字列...][:]	画面への文字表示 文末:で改行の抑制可能
INPUT 変数 INPUT 変数,オーバーフロー時の既定値 INPUT プロンプト, 変数 INPUT プロンプト,変数,オーバーフロー時の既定値	数値の入力 例: INPUT "A=",A,-1
CLS	画面表示内容の全消去
COLOR 文字色 COLOR 文字色,背景色	文字色の設定 ※ターミナル版でのみ利用可能
ATTR 属性	文字表示属性設定 ※ターミナル版でのみ利用可能
LOCATE 横位置,縦位置	カーソルの移動

□ VFD 表示制御関連

コマンド

VCLS	VFD の表示内容の全消去
VDISPLAY モード	VFD の点灯のオン・オフ設定
VBRIGHT 輝度	VFD の輝度設定
VMSG 速度,メッセージ文	VFD にメッセージ文の表示
VSCROLL スクロール幅,速度	VFD に空白挿入表示 (スクロール)
VPUT データ格納アドレス,データ長,速度	VFD にデータ直接出力

□ サウンド関連

コマンド

TONE 周波数,出力期間	指定周波数のパルス出力
NOTONE	パルス出力の停止
PLAY “MML 文”	MML による単音音楽演奏

□ GPIO・入出力関連

コマンド

GPIO ピン名 ピン番号,機能名	GPIO 機能設定 ピン名利用時は GPIO 記述省略可能
OUT ピン番号,出力値	デジタル出力
POUT ピン番号, デューティ値 POUT ピン番号, デューティ値,周波数	PWM パルス出力

数値関数

IN (ピン番号)	デジタル入力
ANA (ピン番号)	アナログ入力
I2CR (デバイスアドレス,コマンドアドレス,コマンド長, データアドレス,データ長)	I2C スレーブデバイスからのデータ受信
I2CW (デバイスアドレス,コマンドアドレス,コマンド長, データアドレス,データ長)	I2C スレーブデバイスへのデータ送信

システムコマンド、一般コマンドの詳細については「7 各コマンド・関数」を参照して下さい。

4.6 数値

□ 数値

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」で使える数値は整数型のみとなります。
整数型は 16 ビット幅、有効範囲は-32768～32767 となります。式、数値定数、数値関数、変数、配列変数はすべてこれに従います。

数値の表記は 10 進数、16 進数の表記可能です。

10 進数表記(-32768～32767) : -1, -32767, 100

16 進数表記(1 桁～4 桁) : \$1345, \$abcd, \$Abcd

(注意)

数値を中間コードに変換（文字列数値を 2 バイト型数値に変換）する際、オーバーフローが発生した場合は overflow エラーとなります。

例①：

```
?32768
Overflow
```

評価・演算においてオーバーフローが発生した場合はエラーとはなりません。

例②：

```
?32767+1
-32768
```

例①、例②は一見、同じような記述ですが、結果に差異が発生することをご理解下さい。

4.7 文字・文字列

□ 文字・文字列

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」では全角シフト JIS、半角カタカナ、半角英数字の文字列を扱うことができます。

文字列の表記は次の通りです。

“文字列” : ダブルクォーテーションで囲みます。
文字列は 0～255 文字まで指定可能です。

例:

```
?PRINT "Hello こんにちは"
Hello こんにちは
OK
```

4.8 文字列利用における制約

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」では制約付きで文字列の利用をサポートしています。
変数への代入も可能です。

例:

```
10 A="ABCDE"
20 PRINT A
RUN
ABCDE
OK
```

変数への文字列代入は、文字列格納アドレスを変数に代入することにより実現しています。

C 言語のポインタによる参照方式と同等です。

ただし、文字列情報を[長さ+文字列]の形式で管理しているため、文字列の参照を代入した変数の値を変更して利用することは行えません。

下記のような利用は出来ません。行った場合、意図しない動作となります。

例：

```
10 A="ABCDE"
20 A=A+1
30 PRINT A
```

また、コマンドラインでの利用は行えません。文字列参照が正しく行うことが出来ません。

例：

```
A="ABCDE"
OK
?A
```

変数に代入した文字列の操作を行う次の関数を用意しています。

- LEN() 文字列の長さの取得

```
10 A="ABCDE"
20 L=LEN(A)
RUN
OK
```

- STR\$() 変数で参照している文字列の取得、部分切り出し

```
10 A="ABCDE"
20 PRINT STR$(A)
30 PRINT STR$(A,4,1)
RUN
ABCDE
D
OK
```

- ASC() 指定位置の文字コードの取得

```
10 A="ABCDE"
20 C=ASC(A,4,1)
RUN
68
OK
```

4.9 変数・配列変数

□ 変数

変数とは数値を格納する保存領域です。数値と同様に式に利用できます。

変数には、通常の変数の他に**配列変数**があります

変数に格納する数値は 16 ビット幅、有効範囲は-32768～32767 となります。

変数は数値型のみ利用可能ですが、文字列の格納アドレスを代入することにより、文字列の参照を行うことが出来ます。

変数の表記

通常の変数：変数名は英字 A～Z の 1 文字、または英字+数字(0～6)の 2 文字の利用が可能です。

例:

```
10 A0=200
20 A1=300
30 A=A0+A1
40 S="Hello"
```

配列変数：@(添え字)の形式で添え字は数値で 0～99(@ (0)～@ (99)) まで利用可能

添え字の数値には式、変数等の利用が可能

例:

```

10 A=5
20 @(0)=30/5,
30 @(A+1)=5

```

代入式において、指定した添え字を起点として連続代入が可能

例:

```

10 @(10)=100,200,300,400,500

```

4.10 演算子

□ 演算子

数値演算で利用できる演算子を示します。

記述例の A,B には数値、変数、配列変数、関数、カッコで囲んだ式が利用できます。

算術演算子

演算子	説明	記述例
+	足し算	A+B A と B を足す
-	引き算	A-B A から B を引く
*	掛け算	A*B A と B の積
/	割り算	A/B A を B で割る
%	剰余算	A%B A を B で割った余り

ビット演算子

演算子	説明	記述例
&	ビット毎の AND 演算	A&B A と B のビット毎の AND
	ビット毎の OR 演算	A B A と B のビット毎の OR
>>	ビット右シフト演算	A>>B A を右に B ビットシフト
<<	ビット左シフト演算	A<<B A を左に B ビットシフト
~	ビット毎の反転	~A A の各ビットを反転
^	ビット毎の XOR	A^B A と B の XOR

比較演算、論理演算の論理反転は、0 が偽、0 以外が真となります。

0 以外が真となりますので、1、-1 はとも真となります。

比較演算子

演算子	説明	記述例
=	等しいかを判定	A=B A と B が等しければ真, 異なれば偽
!=	異なるかを判定	A!=B A と B が異なれば真, 等しければ偽
<>		
<	小さいかを判定	A<B A が B 未満であれば真、そうでなければ偽
<=	小さいまたは等しいかを判定	A<=B A が B 以下であれば真、そうでなければ偽
>	大きいかを判定	A>B A が B より大きければ真、そうでなければ偽
>=	大きいまたは等しいかを判定	A>=B A が B 以上であれば真、そうでなければ偽

論理演算子

演算子	説明	記述例
AND	論理積	A AND B A,B が真なら真、でなければ偽
OR	論理和	A OR B A,B どちらかが真なら真、でなければ偽
!	否定	!A A が真なら偽、偽なら真

□ 演算子の優先順序

演算子の優先度を下記に示します。優先度の数値が小さいほど優先度が高くなります。

計算結果が意図した結果にならない場合は、優先度の確認、括弧をつけて優先度を上げる等の対応を行って下さい。

演算子の優先度

優先度	演算子
1	括弧で囲った式
2	!, ~
3	*, / , % , &, . << , >>, ^
4	+, - ,
5	=, <> , != , > , >= , < , <= , AND , OR

比較演算子と論理演算は優先度が同レベルです。比較演算子の演算を先に行う場合は、括弧を使って優先度を上げて下さい。

例：

```
?1<5 and 2>3
0
?(1<5) and (2>3)
1
```

4.11 式

□ 式

式とは 1、1+1、A、A+1、ABS(-1) などの演算・値の評価を伴う記述をいいます。

式はプログラム実行にて計算・評価が行われ、1つの整数値の値として振る舞います。

変数への値の代入、コマンド、条件判定(IF 文)、関数に引数に式が用いられた場合は、評価後の値がコマンドおよび関数に渡されます。

式の利用：

変数への値の代入(代入命令 LET は省略可能)

```
LET A=(1+10)*10/2
A=5*5+3*2
@(I+J) = I*J
```

コマンド・関数の引数

```
LOCATE X+I*2, J:PRINT "*"
OUT PC13, I>J
A=EEPREAD(I+J+1)/2
```

4.12 定数

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」では次の定数が利用出来ます。

定数はコマンドの引数や式の中で数値関数と同等に利用出来ます。

□ 1 ビット入出力値

HIGH, LOW

HIGH は 1, LOW は 0 が割り当てられています。

各コマンドの引数、IF 文、式にて利用出来ます。

□ 仮想メモリ領域先頭アドレス定数

VRAM,VAR,ARRAY,PRG,MEM,FNT,GRAM

データ領域を参照するための定数です。各定数の詳細は次の通りです。

定数名	値	領域サイズ	用途
VRAM	\$0000	3,840	スクリーン表示用メモリ (CW×CH)
VAR	\$1000	440	変数領域 (A~Z, A0~Z6) + 4 バイト
ARRAY	\$1240	200	配列変数領域 (@ (0) ~ @ (99))
PRG	\$1340	4,096	プログラム領域
MEM	\$2340	1,024	ユーザーワーク領域
FNT	\$2740	1,283	フォント 256 文字
GRAM	\$37A0	27	LED マトリックス表示用メモリ
FNT2	\$38A0	13,680	美咲フォント 1710 字

定数値のアドレスは仮想的なアドレスです。実アドレスとは異なります。

PEEK、POKE、I2CW、I2CR、BITMAP、LDBMP のメモリ領域を利用するコマンドで利用できます。

定数の利用例：仮想メモリ領域先頭アドレス定数を表示する

```
10 'カメラリポート' リス
20 PRINT "カメラリポート' リス"
30 PRINT "VRAM  ",HEX$(VRAM,4)
40 PRINT "VAR   ",HEX$(VAR,4)
50 PRINT "ARRAY:",HEX$(ARRAY,4)
60 PRINT "PRG   :",HEX$(PRG,4)
70 PRINT "MEM   :",HEX$(MEM,4)
80 PRINT "FNT   :",HEX$(FNT,4)
90 PRINT "GRAM  :",HEX$(GRAM,4)
100 PRINT "FNT2  :",HEX$(FNT2,4)
```

実行結果

```
run
カメラリポート' リス
VRAM 0000
VAR 1900
ARRAY:1AA0
PRG :1BA0
MEM :2BA0
FNT :2FA0
GRAM :37A0
FNT2 :38A0
OK
```

□ 画面表示の定数

画面サイズを参照する定数です。

CW : キャラクタ画面の横文字数 (デフォルト 80)
 CH : キャラクタ画面の縦行数 (デフォルト 24)
 GW : LED マトリックスの横ドット数(5)
 GH : LED マトリックスの縦ドット数(5)

□ 方向を示す定数

GSCROLL、MSG コマンドでスクロール方向を指定に利用出来ます。

UP : 値 0 上方向
 DOWN: 値 1 下方向
 RIGHT: 値 2 右方向
 LEFT : 値 3 左方向
 TOP : 値 4 上面(スクロールなし)

□ ピン番号定数

GPIO、OUT、IN、ANA、SHIFTOUT、SHIFTIN コマンド・関数のピン番号の指定に利用します。

各ピン番号定数に実際のピン番号 0 ~ 32 が割り当てられています。

PNO ~ PN32

□ GPIO モード設定定数

GPIO コマンドのモード設定を行うための定数です。

OUTPUT, INPUT_PU, INPUT_PD, INPUT_FL

□ ビット方向定数

LSB, MSB

ビット送信等で上位、下位を指定するための定数です。

SHIFTIN, SHFITOUT コマンドで利用します。

□ OFF/ON 定数

OFF : 値 0

ON : 値 1

□ キー入力コード定数

INKEY()関数によるキーボードからの入力判定等に利用するための定数です。

KENTER : 13

KUP : 28

KDOWN : 29

KRIGHT : 30

KLEFT : 31

KSPASE : 32

5. 各制御文の詳細

5.1 GOTO 指定行へのジャンプ（制御命令）

□ 書式

GOTO 行番号

GOTO "ラベル"

□ 説明

プログラムの実行を行番号、"ラベル"で指定した行にジャンプします。

行番号には数値、式（変数、関数を含む）が利用可能です。

ラベルを指定した場合、行先頭に同じラベルがある行にジャンプします。

ラベルはダブルクォテーション(")で囲って指定します。

```
10 I=0
20 "LOOP"
30 PRINT "@"
40 I=I+1:IF I=5 GOTO 60
50 GOTO "LOOP"
60 END
```

上記の例では、40 行の GOTO で 60 行にジャンプ、50 行のラベル指定で 20 行にジャンプしています。

ラベルを使うとプログラムの流れがわかりやすくなります。

行の先頭に無いラベルは、ジャンプ先としての利用は出来ません。

```
10 GOTO "hoge"
20 END
30 ? "Hello": "hoge": ? "test"
run
Undefined line number or label in 10
10 GOTO "hoge"
OK
```

行番号には式や変数の指定が可能ですが、RENUM コマンドによる行番号付け替えに対応出来ない場合があります。

①GOTO N*100+500

②GOTO 500+N*100+500

①では RENUM コマンドは何もしません。②は 500 を番号とみなして更新します。

ただし①②とも計算結果が正しい行番号となることは保証出来ません。ご注意下さい。

□ エラーメッセージ

Undefined line number or label	: 指定した行、ラベルが存在しない
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
Syntax error	: 文法エラー、書式と異なる利用を行った

□ 利用例

ラベルを使ったループ処理

```
100 GOSUB "SUB01"
110 GOSUB "SUB02"
120 N=0
130 "LOOP"
140 PRINT "N=";N
150 N=N+1:IF N<5 GOTO "LOOP"
160 END
170 "SUB01"
180 PRINT "SUB01"
190 RETURN
200 "SUB02"
210 PRINT "SUB02"
220 RETURN
```

5.2 GOSUB サブルーチンの呼び出し（制御命令）

□ 書式

GOSUB 行番号
GOSUB “ラベル”

□ 説明

プログラムの実行を一旦、行番号またはラベルで指定した行に移ります。
移った先で RETURN 命令により、呼び出した GOSUB 命令の次の位置に戻ります。
行番号には数値、式（変数、関数を含む）が利用できます。
ラベルを指定した場合は、行先頭に同じラベルがある行に移ります。
ラベルはダブルクォテーション(“)で囲って指定します。

```
10 GOSUB "PRN_LOGO"
20 GOSUB "PRN_DATE"
30 GOSUB 200
40 END
50 "PRN_LOGO"
60 PRINT "Tiny BASIC forArduino"
70 RETURN
80 "PRN_DATE"
90 PRINT "Edition V0.01 2018/01/16"
100 RETURN
200 PRINT "Ready"
210 RETURN
```

上記の例では、10 行、20 行でラベルを指定してサブルーチンを呼び出しています。
30 行では、行番号を指定してサブルーチンを呼び出しています。
ラベルを使うことで、プログラムの実行がわかりやすくなります。
行の先頭に無いラベルは、GOSUB 文の呼び出し先としての参照はされません。

```
10 GOSUB "hoge"
20 END
30 ? "Hello":"hoge":?"test":RETURN
run
Undefined line number or label in 10
10 GOSUB "hoge"
OK
```

行番号には式や変数の指定が可能です。RENUM コマンドによる行番号付け替えに対応出来ない場合があります。

①GOSUB N*100+500

②GOSUB 500+N*100+500

①では RENUM コマンドは何もしません。②は 500 を行番号とみなして更新します。
ただし①②とも計算結果が正しい行番号となることは保証出来ません。ご注意ください。

□ エラーメッセージ

Undefined line number or label	: 指定した行、ラベルが存在しない
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
GOSUB too many nested	: GOSUB のネスト数が規定(=10)を超えた
Syntax error	: 文法エラー、書式と異なる利用を行った

□ 利用例

ラベルを使ったサブルーチンの呼び出し例

```
100 GOSUB "SUB01"
110 GOSUB "SUB02"
120 N=0
130 "LOOP"
140 PRINT "N=";N
150 N=N+1:IF N<5 GOTO "LOOP"
160 END
170 "SUB01":PRINT "SUB01":RETURN
200 "SUB02":PRINT "SUB02":RETURN
```


5.3 RETURN GOSUB 呼び出し元への復帰（制御命令）

□ 書式

RETURN

□ 説明

直前に呼び出された GOSUB の次の処理に復帰します。

詳細は「6.2 GOSUB サブルーチンの呼び出し（制御命令）」を参照して下さい。

□ エラーメッセージ

Undefined line number or label	: 指定した行、ラベルが存在しない
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
RETURN stack underflow	: GOSUB の呼び出しがないのに RETURN を実行
Syntax error	: 文法エラー、書式と異なる利用を行った

□ 利用例

ラベルを使ったサブルーチンの呼び出し例

```

100 GOSUB "SUB01"
110 GOSUB "SUB02"
120 N=0
130 "LOOP"
140 PRINT "N=";N
150 N=N+1:IF N<5 GOTO "LOOP"
160 END
170 "SUB01"
180 PRINT "SUB01"
190 RETURN
200 "SUB02"
210 PRINT "SUB02"
220 RETURN

```

5.4 IF 条件判定（制御命令）

□ 書式

IF 条件式 実行文 1

IF 条件式 実行文 1 ELSE 実行文 2

□ 説明

条件式の真偽判定を行い、真の場合は実行文 1 を実行します。

真の場合に ELSE がある場合、ELSE 文以降の命令文は実行しません。

偽の場合、ELSE がある場合、実行文 2 を実行します。なければ次の行にスキップします。

真偽は次の条件となります。

真：評価した値が 0 以外である

偽：評価した値が 0 である

真偽判定を行う条件式には定数、数値、関数を含む計算式が記述できます。

式の例：

```
IF A > 5 B=B-1
IF A+1 PRINT A
IF A & 1 TONE 440,200
IF INKEY() = ASC("A") GOTO 100
```

定数の例

```
IF HIGH GOTO 100
```

数値の例

```
IF 123 PRINT "123"
```

関数の例

```
IF RND(1) X=X+1
```

実行文には IF、GOTO、GOSUB 等の制御命令を使うことも可能です。

（注意）ELSE 利用の制約(ぶら下がり ELSE に関する補足)

IF 文をネストして利用した場合、ELSE 文は直前の IF 文に対応します。

例：IF A=1 IF B=1 ? "A,B=1" ELSE ? "A=1,B<>1"

上記の ELSE は 2 番目の IF 文に対応します。

この「ELSE 文は直前の IF 文に対応」の条件で、ELSE 文に IF 文をネスト出来ます。

例: IF A=1 IF B=1 ? "A,B=1" ELSE IF B=2 ? "A=1,B=2" ELSE IF B=3 ? "A=1,B=3"

上記では、A=1,B=1,2,3 の条件に対して対応する表示メッセージを表示する例です。

□ エラーメッセージ

IF without condition : 条件式が定義されていない

Overflow : 指定した数値が -32768 ~ 32767 を超えている

□ 利用例

y を押したら、“Y key” を表示して終了、その他のキーなら“End” を表示して終了する

何も押されていないならば 10 行から 30 行をループする。

```
10 A=INKEY()
20 IF A=ASC("y") PRINT "Y key":END ELSE IF A<>0 PRINT "End":END
30 GOTO 10
```

5.5 FOR TO ~ NEXT 繰り返し実行（制御命令）

□ 書式

FOR 変数=初期値 TO 最終値 実行文(複数行可能) NEXT 変数

FOR 変数=初期値 TO 最終値 STEP 増分 繰り返し実行文(複数行可能) NEXT 変数

□ 説明

変数を初期値から最終値まで増やし、FOR から NEXT の間の実行文を繰り返し実行します。

変数が最終値に達した時点で繰り返しを止め、NEXT の次の命令の実行を行います。

STEP にて増分を指定しない場合、増分は 1 となります。

STEP を用いた場合は、マイナス値を含め任意の増分の指定が可能です。

例：

```
10 PRINT "start."
20 FOR I=0 TO 5 STEP 2
30 PRINT I
40 NEXT I
50 PRINT "done."
```

実行結果

```
start
0
2
4
done.
ok
```

上記の例では I を 0 から 5 まで、2 ずつ増加して 30 行の命令を繰り返し実行します。

I が 4 の時、増分 2 を足すと終了値 5 を超えた 6 となるので繰り返しを終了し、50 行を実行します。

FOR 文はネストも可能です(利用例を参照)。

FOR~NEXT の繰り返しは、利用している変数が繰り返し条件を満たさなくなった時点で繰り返しを終了します。

最初の例に“35 I=6”を追加し、ループ内で条件を満たさなくすることでループを抜けることが出来ます。

```
10 PRINT "start."
20 FOR I=0 TO 5 STEP 2
30 PRINT I
35 I=6
40 NEXT I
50 PRINT "done."
```

実行結果

```
start
0
done.
ok
```

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
FOR without variable	: FOR 文で変数を指定していない
FOR without TO	: FOR 文で TO を指定していない
NEXT without counter	: NEXT に対応する FOR 文が無い
FOR too many nested	: FOR 文のネスト数が規定数(=10)を超えた
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

画面の指定位置に*を表示する（FOR 文のネストの例）

```
10 CLS
20 FOR Y=5 TO 10
30 FOR X=5 TO 10
40 LOCATE X,Y:PRINT "*"
50 NEXT X
60 NEXT Y
```

5.6 END プログラムの終了（制御命令）

□ 書式

END

□ 説明

プログラムの実行を終了します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

□ 利用例

特定の条件でプログラムを終了する

```
10 I=0  
20 PRINT I  
30 IF I=5 END  
40 GOTO 20
```

6. 各コマンド・関数の詳細

6.1 RUN プログラムの実行（システムコマンド）

□ 書式

RUN

□ 引数

なし

□ 説明

プログラムの実行を行います。

実行中のプログラムは [ESC]キーまたは、[CTRL+C]キーにて強制終了することが出来ます。

シリアルコンソール上で[ESC]キーにて中断する場合は、[ESC]キーを2回押す必要があります。

プログラムの実行中は、カーソルが非表示となります。

プログラムの実行中は実行すべき次の行が無い場合、実行を終了します。

また END コマンドにより実行を終了します。

プログラムの実行中にエラーが発生した場合は実行を終了します。

（注意） WAIT 命令で長い時間待ちを行っている場合は、時間待ちが終了するまで、[ESC]キー、[CTRL+C]キーによる強制終了を行うことが出来ません。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

□ 利用例

プログラムを実行する

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK

RUN
*****
OK
```

6.2 RENUM 行番号の振り直し（システムコマンド）

□ 書式

RENUM

RENUM 開始行番号

RENUM 開始行番号,増分

□ 引数

開始番号 : 振り直しをする新しい行番号の開始番号 (0 ~ 32767)

増分 : 行番号の増分 (1 ~ 32767)

□ 説明

プログラム全体の行番号を指定した条件にて振り直します。

引数を省略した場合は、行番号を 10 行から 10 間隔で振り直します。

開始番号のみ指定した場合、指定した開始番号から 10 間隔で振り直します。

開始番号と増分を指定した場合、指定した開始番号から指定した増分で振り直します。

振り直しにおいて、GOTO 文、GOSUB 文のとび先の行番号も更新されます。

(注意) GOTO,GOSUB に存在しない行番号を指定している場合、更新は行われません。
また、行番号に計算式を利用している場合、正しい更新が行われない場合があります。
GOTO 100+N*10
の記載の場合、先頭の数値 100 を行番号とみなして更新します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

Illegal value : 振り直しをする新しい行番号が有効範囲を超えている

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

行番号を 100 から 10 間隔で振り直す

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK
RENUM 100,10
OK
LIST
100 I=0
110 PRINT "*";
120 I=I+1:IF I<10 GOTO 110
130 PRINT
140 END
```

6.3 LIST プログラムリストの表示

□ 書式

LIST

LIST 表示開始行番号

LIST 表示開始行番号, 表示終了行番号

□ 引数

表示開始行番号：表示を開始する行番号（0 ～ 32767）

表示終了行番号：表示を終了する行番号（0 ～ 32767）

□ 説明

プログラムリストの表示を行います。

引数を指定しない場合は、全てのプログラムを表示します。

表示開始番号を指定した場合は、その番号以降のプログラムリストを表示します。

表示開始番号、表示終了番号を指定した場合は、その範囲のプログラムリストを表示します。

表示したプログラムリストは、カーソルを移動して編集することが出来ます。

編集後は必ず[ENTER]キーを押して入力確定を行って下さい。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

□ 利用例

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK

list 20,30
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
OK
```

6.4 NEW プログラムの消去

□ 書式

NEW

□ 引数

なし

□ 説明

プログラム領域のプログラム、変数、配列変数を消去します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した
 Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

プログラムを消去する

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK

run
*****
OK
?I
10
OK
NEW
OK
LIST
OK
?I
0
OK
```


6.5 LOAD 内部 EEPROM メモリからプログラムを読み（システムコマンド）

□ 書式

LOAD

LOAD プログラム番号

□ 引数

プログラム番号 : 0 ~ 1 内部 EEPROM メモリのプログラム番号

□ 説明

内部フラッシュメモリからプログラムを読み込みます。

引数を省略した場合は、内部フラッシュメモリのプログラム番号を読み込みます。

引数にプログラム番号を指定した場合は、指定したプログラム番号を読み込みます。

プログラム番号 0~1 で指定します。読込の際、変数領域は初期化されます。

□ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した
 Illegal value : プログラム番号の指定が 0~1 の範囲外である
 Program not found : 指定したプログラム番号にプログラムが保存されていない
 Overflow : 指定した数値が-32767 ~ 32767 を超えている

□ 利用例

プログラム番号 1 を読み込む

LOAD 1
OK

6.6 SAVE 内部 EEPROM メモリへのプログラム保存

□ 書式

SAVE

SAVE プログラム番号

□ 引数

プログラム番号 : 0 ~ 1

□ 説明

プログラムをマイコン内のフラッシュメモリに保存します。

引数の省略、数値を指定した場合は内部フラッシュメモリに保存します。

プログラムは最大で 2 つ保存可能です。保存先はプログラム番号 0~7 で指定します。

引数の省略した場合、プログラム番号 0 に保存します。

□ エラーメッセージ

Syntax error	: 書式と異なる利用を行った、プログラム番号に変数、式を指定した
Illegal value	: プログラム番号の指定が 0~1 の範囲外である
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

プログラム番号 1 にプログラムを保存する

SAVE 1
OK

6.7 FILES 内部フ EEPROM メモリ内保存プログラムの一覧表示

□ 書式

FILES

FILES 開始プログラム番号

FILES 開始プログラム番号,終了プログラム番号

□ 引数

開始プログラム番号 : 表示開始プログラム番号 0~1

終了プログラム番号 : 表示終了プログラム番号 0~1

□ 説明

内部 EEPROM メモリに保存されているプログラムの一覧を表示します。

引数を指定しない場合は、内部フラッシュメモリ内のプログラム番号 0~1 の先頭行をリスト表示します。

開始プログラム番号,終了プログラム番号を指定した場合、その範囲のリストを表示します。

プログラム番号にプログラムが保存されていない場合は(none)と表示されます。

プログラム先頭行にコメントをつけると、一覧表示でのプログラムの内容が分かり易くなります。

例:

```
files
0: 'VDF表示サンプル
1: 'VFDジグザグ表示
OK
```

(注意) 豊四季タイニーBASIC (ファームウェア) の新規利用または更新を行った直後は、EEPROM メモリ上の既存データの内容の不整合により正しく表示できない場合があります。その場合は、ERASE コマンドにてフラッシュメモリ上のプログラムの消去を行って下さい。

□ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した
 Illegal value : 指定した開始プログラム番号,終了プログラム番号の値が正しくない。
 Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

プログラムを作成し、先頭行にコメントを付けて保存します。

保存しているプログラムを FILE コマンドで確認します。

コメントには日本語(シフト JIS) の記述も可能です。

```
10 'こんにちは、世界
20 ?"こんにちは、世界"
save
OK
files
0: 'こんにちは、世界
1: (none)
OK
run
こんにちは、世界
OK
```

6.8 ERASE 内部フラッシュメモリ上のプログラム削除

□ 書式

ERASE プログラム番号

ERASE 開始プログラム番号,終了プログラム番号

□ 引数

プログラム番号 : 0 ~ 1

開始プログラム番号 : 0 ~ 1

終了プログラム番号 : 0 ~ 1

□ 説明

内部 EEPROM 保存されているプログラム(プログラム番号 0~1)の削除を行います。

削除対象プログラムは、コマンド引数にてプログラム番号を指定するか、

削除するプログラム番号の範囲（開始プログラム番号、終了プログラム番号）にて指定します。

(注意) 豊四季 Tiny BASIC のスケッチを書き込み後、FILES コマンドにてプログラム一覧を表示した場合、意味不明の文字列が大量に表示される場合があります。

この場合は ERASE コマンドを実行して、プログラム保存領域を初期化して下さい。

□ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Illegal value : プログラム番号の指定が 0~7 の範囲外である

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

フラッシュメモリに保存されている全てのプログラムの削除を行う。

```
ERASE 0,1
OK
FILES
0: (none)
1: (none)
OK
```

6.9 EDIT スクリーンエディタ利用設定 (システムコマンド)

□ 書式

EDIT モード

□ 引数

モード : OFF スクリーンエディタ有効、ON スクリーンエディタ無効 (デフォルト)
 または 0 スクリーンエディタ有効、1 スクリーンエディタ無効

□ 説明

スクリーンエディタ機能の有効、無効設定をおこないます。
 デフォルトでは、スクリーンエディタは有効になっています。

スクリーンエディタは有効な状態では、編集キーを使った編集機能を利用することが出来ます。
 利用可能な編集キーについては「1.10 編集操作操作キー」を参照して下さい。

「豊四季タイニーBASIC for MW25616L 実験用表示モジュール」では、SRAM の容量の制約により、スクリーンエディタの画面サイズは、横 48 文字、縦 8 行となります。
 スクリーンエディタ有効時は、8 行の範囲でスクロール表示されます。

スクリーンエディタ機能を無効にした場合は行単位の入力となります。
 その変わり、スクロール範囲はターミナル・ソフトのウィンドウの範囲となり、長いリストの閲覧を楽に行うことが出来ます。ただし、カーソル制御を行う LOCATE コマンド等の機能の一部が無効となる等の制約が発生します。
 スクリーンエディタ機能を無効の状態では、行カーソル`>`が表示されます。

```
> files
0:(none)
1:(none)
OK
>
```

(注意) モード切替直後は、表示が乱れたり、カーソルが正しい位置に表示されない場合があります。
 その場合は、CLS コマンドの実行と ENTER キーの入力を試して下さい。

□ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した
 Illegal value : プログラム番号の指定が 0~7 の範囲外である
 Overflow : 指定した数値が -32768 ~ 32767 を超えている

□ 利用例

スクリーンエディタの利用を無効にする。

```
EDIT OFF
OK
>
```


6.10 FREE プログラム領域の残りバイト数の取得（数値関数）

□ 書式

FREE()

□ 引数

なし

□ 戻り値

プログラム領域の残りバイト数 0 ～ 512

□ 説明

プログラム領域の残りバイト数を返します。

作成したプログラムサイズを確認する場合は、下記の記述にて行うことが可能です。

```
PRINT 512-FREE()
```

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : 引数部の記述が正しくない

□ 利用例

プログラムサイズを調べる

```
PRINT 512-FREE()
287
OK
```

6.11 REM コメント

□ 書式

REM コメント

‘ コメント

□ 引数

コメント : 任意の文字列

□ 説明

プログラムに説明等の記載を行います。

‘(シングルクォート)は REM の省略形です。

REM および以降の文字以降はすべてコメントとみなし、プログラムとして実行されません。

プログラムの先頭行にコメントを付けた場合は、FILES コマンドで保存プログラム一覧を表示時に、各プログラムの見出しとなります。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

□ 利用例

コメントの記述例

```
LIST
1 REM Print starts
10 I=0:'initialize value
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
```


6.12 LET 変数に値を代入

□ 書式

LET 変数=値

LET 配列変数=値,値,値,値,...

変数=値

配列変数=値,値,値,値,...

□ 引数

変数 : 変数、または配列変数

配列変数 : 配列変数 @(値)

配列の値(添え字)には式、数値、変数、配列変数、数値定数の利用が可能

値 : 式、数値、変数、配列変数、数値定数

□ 説明

変数に値を代入します。

値は式、数値、定数、変数、配列変数等を評価した整数値です。

配列変数への代入は、複数の値を指定した連続代入が可能です。指定した添え字を起点に順番に代入します。

LET @(3)=1,2,3,4,5

上記の記述は、

LET @(3)=1: LET @(4)=2: LET @(5)=3: LET @(5)=4: LET @(5)=5

と同じです。

LET は省略可能です。次の 2 は同じ結果となります。

LET A=A+1

A=A+1

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

```

LIST
1 'Print starts
10 LET I=0:'initialize value
20 PRINT "*";
30 LET I=I+1:IF I<10 GOTO 20
40 PRINT
50 END

```

6.13 ABS 絶対値の取得（数値関数）

□ 書式

ABS(値)

□ 引数

値 : -32767 ~ 32767

式、変数、配列変数、数値、数値定数の指定が可能

□ 戻り値

指定した値の絶対値

□ 説明

指定した値の絶対値を返します。

（注意）-32768 の絶対値 32768 はオーバーフローとなるため、-32768 を指定した場合はオーバーフローエラーとなります。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Overflow : 指定した数値が-32767 ~ 32767 を超えている
 '(' or ')' expected : '(' または ')' が無い

□ 利用例

変数の値の絶対値を表示する

```
10 FOR I=-3 TO 3
20 PRINT ABS(I)
30 NEXT I
```

RUN

```
3
2
1
0
2
3
OK
```

6.14 ASC 文字から文字コードへの変換（数値関数）

□ 書式

ASC(文字列)

ASC(文字列,文字位置)

ASC(変数)

ASC(変数,文字位置)

□ 引数

文字列： “文字列”
 “ABC” の形式とし、ダブルクォーテーション文字を囲みます

文字位置： 1～32767
 変換対象となる左からの文字位置を指定します。

変数： 文字列参照している変数または配列変数

□ 戻り値

指定文字に対応する文字コード（0 ～ 255）

□ 説明

指定した半角文字に対応する文字コードを返します。

2 文字以上の文字列を指定した場合、先頭の文字コードを返します。

```
10 ?ASC("ABCD")
RUN
65
OK
```

文字位置を指定した場合、指定した位置の文字コードを返します。

（全角文字が含まれている場合、全角 1 文字は 2 文字として扱います）

```
10 ?ASC("ABCD",3)
RUN
67
OK
```

変数を指定した場合、変数が参照している文字列のコードを返します。

```
10 A="ABCDEF"
20 ?ASC(A)
30 ?ASC(A,3)
RUN
65
67
OK
```

（注意） 変数の文字列参照はプログラム中にのみ有効です。コマンドラインでは文字列参照を正しく行うことが出来ません。

（注意） 全角文字の文字コードを取得する場合は WASC コマンドを利用して下さい。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 文字指定位置が不当

'(' or ')' expected : '(' または ')' がない

Overflow : 指定した数値が -32768 ～ 32767 を超えている

□ 利用例

アルファベット A～Z を表示する

```
10 C=ASC("A")
20 FOR I=0 TO 25
30 PRINT CHR$(C+I);
40 NEXT I
50 PRINT
```

```
run
ABCDEFGHIJKLMNOPQRSTUVWXYZ
OK
```

6.15 WASC 全角を含む文字から文字コードへの変換（数値関数）

□ 書式

WASC(文字列)

WASC(文字列,文字位置)

WASC(変数)

WASC(変数,文字位置)

□ 引数

文字列： “文字列”
 “あいう ABC” の形式とし、ダブルクォーテーション文字を囲みます

文字位置： 1～32767
 変換対象となる左からの文字位置を指定します。

変数： 文字列参照している変数または配列変数

□ 戻り値

指定文字に対応するシフト JIS コード (0 ～ \$FFFF)

□ 説明

指定した全角文字または半角文字に対応する文字コードを返します。

半角文字の場合は、ASC()と同じ結果となります。

2 文字以上の文字列を指定した場合、先頭の文字コードを返します。

```
10 ?HEX$(WASC("あ ABCD"))
RUN
82A0
OK
```

文字位置を指定した場合、指定した位置の文字コードを返します。

全角 1 文字は 1 文字とカウントします。次の例では 4 番目の文字“う”の文字コードを取得しています。

```
10 ?HEX$(WASC("A あ B う CD",4))
RUN
82A4
OK
```

変数を指定した場合、変数が参照している文字列のコードを返します。

```
10 A="AB あ CDEF"
20 ?HEX$(WASC(A))
30 ?HEX$(WASC(A,3))
RUN
41
82A0
OK
```

(注意) 変数の文字列参照はプログラム中にのみ有効です。コマンドラインでは文字列参照を正しく行うことが出来ません。

WASC()関数は全角文字を 1 文字として処理します。

ASC()全角文字を 2 バイトとして処理します。

半角のみの文字列を扱う場合は、ASC()を利用して下さい。WASC()関数は全角文字判定処理を行っているため、ASC()よりも処理に時間がかかります。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 文字指定位置が不当
'(' or ')' expected	: '(' または ')'がない
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

文字列に該当するフォントデータを表示します。

※このサンプルプログラムは、MW25616L 実験用表示モジュールでは未サポートです。

```

10 S="あ"
20 A=WADR(WASC(S))
30 FOR Y=0 TO 7
40 D=PEEK(A+Y)
50 FOR X=0 TO 7
60 IF D&($80>>X) ?"■"; ELSE ?" ";
70 NEXT X
80 ?
90 NEXT Y

```

実行結果

```

run
  ■
 ■■■■■
  ■
   ■■■■
  ■■ ■■
 ■ ■■■ ■■
■ ■■■ ■■
  ■■ ■
OK

```

6.16 LEN 文字列の長さの取得（数値関数）

□ 書式

LEN(変数)

LEN(文字列)

□ 引数

変数 : 変数、または配列変数

文字列 : “文字列”の形式（ダブルクォーテーションで囲み）

□ 戻り値

文字列の長さ 0 ～ 32767

□ 説明

文字列定数、変数で参照してる文字列の文字数をカウントし、その値を返します。

```

10 ?LEN("12345678")
20 A="ABCDEF"
30 @(0)="abcdef"
40 ?LEN(A)
50?LEN(@(0))
RUN
8
6
6
OK

```

（注意） 文字列のサポートは限定的です。
 他の文字列関数と組み合わせた利用はサポートしていません。
 例) 下記のような記述は出来ません。
 ?LEN(BIN\$(100))

本関数は全角 1 文字は 2 文字としてカウントします。全角を 1 文字として扱いたい場合は、WLEN()を利用して下さい。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Overflow : 指定した数値が-32768 ～ 32767 を超えている
 '(' or ')' expected : '(' または ')'が無い

□ 利用例

変数 A が参照している文字列の長さ LEN()関数で取得し、各文字の文字コードを表示します。

```

10 'LEN()サンプル
20 A="Hello!"
30 FOR I=1 TO LEN(A)
40 ?STR$(A,I,1);":";ASC(A,I)
50 NEXT I

```

実行結果

```

run
H:72
e:101
l:108
l:108
o:111
!:33
OK

```

6.17 WLEN 全角を含む文字列の長さの取得（数値関数）

□ 書式

WLEN(変数)

WLEN(文字列)

□ 引数

変数 : 変数、または配列変数

文字列 : “文字列”の形式（ダブルクォーテーションで囲み）

□ 戻り値

文字列の長さ 0 ～ 32767

□ 説明

文字列定数、変数で参照してる文字列の文字数をカウントし、その値を返します。

全角文字、半角文字とも 1 文字としてカウントします。

```

10 ?WLEN("1 あ 2 い 34 う 5678")
20 A="AB さいたま CDEF"
30 @(0)="abcdef 埼玉"
40 ?WLEN(A)
50 ?WLEN(@(0))
RUN
11
10
8
OK

```

（注意）文字列のサポートは限定的です。
 他の文字列関数と組み合わせた利用はサポートしていません。
 例）下記のような記述は出来ません。
 ?WLEN(BIN\$(100))

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Overflow : 指定した数値が-32768 ～ 32767 を超えている
 '(' or ')' expected : '(' または ')'が無い

□ 利用例

変数 A が参照している文字列の長さ WLEN() 関数で取得し、各文字の文字コードを表示します。

```

10 'WLRN()サンプル
20 A="こんにちは世界!"
30 FOR I=1 TO WLEN(A)
40 ?WSTR$(A,I,1);";";HEX$(WASC(A,I))
50 NEXT I

```

実行結果

```

run
こ:82B1
ん:82F1
に:82C9
ち:82BF
は:82CD
世:90A2
界:8A45
!:21
OK

```


6.18 CHR\$ 文字コードから文字への変換（文字列関数）

□ 書式

CHR\$(文字コード)

CHR\$(文字コード, 文字コード, ... 文字コード)

□ 引数

文字コード： 0～255

□ 戻り値

指定した文字コードに対する文字

□ 説明

指定した文字コードに対応する文字を返します。

文字コードは複数指定することが可能です。

PRINT、MSG、GPRINT の引数にて利用可能です。

範囲外の値を指定した場合は空白文字(" ")を返します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 文字コードに 0～255 以外の数値を指定した

'(' or ')' expected : '(' または ')' が無い

Overflow : 指定した数値が -32768 ～ 32767 を超えている

□ 利用例

アルファベット A～Z を表示する

```

10 C=ASC("A")
20 FOR I=0 TO 25
30 PRINT CHR$(C+I);
40 NEXT I
50 PRINT

```

```

run
ABCDEFGHIJKLMNOPQRSTUVWXYZ
OK

```

6.19 WCHR\$ シフト JIS コードから文字への変換（文字列関数）

□ 書式

WCHR\$(文字コード)

WCHR\$(文字コード, 文字コード, ... , 文字コード)

□ 引数

シフト SJIS 文字コード： 0～\$FFFF

□ 戻り値

指定した文字コードに対する文字

□ 説明

指定したシフト JIS コードに対応する文字を返します。 半角コードにも対応します。

文字コードは複数指定することが可能です。

PRINT、MSG、GPRINT の引数にて利用可能です。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 文字コードに 0～255 以外の数値を指定した
'(' or ')' expected	: '(' または ')' が無い
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

□ 利用例

ひらがな“あいうえお”を表示する

10 ?WCHR\$(\$82A0,\$82A2,\$82A4,\$82A6,\$82A8)

```
run
あいうえお
OK
```

6.20 BIN\$ 数値から 2 進数文字列への変換（文字列関数）

□ 書式

BIN\$(数値)

BIN\$(数値 , 桁数)

□ 引数

数値： 変換対象の整数値(-32768 ~ 32767)

桁数： 出力桁数(0 ~ 16)

□ 戻り値

2 進数文字列(1 桁~16 桁)

□ 説明

指定した数値を 2 進数文字列に変換します。

数値には式、変数、定数等の指定が可能です。

PRINT、MSG、GPRINT の引数にて利用可能です。

桁数を指定した場合は数値が指定した桁数に満たない場合は、0 で桁を補います。

指定した桁数を超える場合は数値の桁数を優先します。

桁数を指定しない場合は、先頭の 0 は付加されません。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

Illegal value : 桁数の値が正しくない

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

2 進数で変数の内容を表示する

```

10 A=1234:B=1
20 PRINT BIN$(A)
30 PRINT BIN$(B)
40 PRINT BIN$(A,4)
50 PRINT BIN$(B,4)

```

```

run
10011010010
1
10011010010
0001
OK

```

6.21 HEX\$ 数値から 16 進数文字列への変換（文字列関数）

□ 書式

HEX\$(数値)

HEX\$(数値 , 桁数)

□ 引数

数値： 変換対象の整数値(-32767 ~ 32767)

桁数： 出力桁数(0 ~ 4)

□ 戻り値

16 進数文字列(1 桁~4 桁)

□ 説明

指定した数値を 16 進数文字列に変換します。

数値には式、変数、定数等の指定が可能です。

PRINT、MSG、GPRINT の引数にて利用可能です。

桁数を指定した場合は数値が指定した桁数に満たない場合は、0 で桁を補います。

指定した桁数を超える場合は数値の桁数を優先します。

桁数をしない場合は、先頭の 0 は付加されません。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

Illegal value : 桁数の値が正しくない

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

入力した整数値を 16 進数表示する

```
10 INPUT "value=",V
20 PRINT HEX$(V,4)
30 GOTO 10
```

6.22 STR\$ 変数が参照する文字列の取得・文字列の切り出し

□ 書式

STR\$(変数)

STR\$(変数,先頭位置,長さ)

STR\$(文字列)

STR\$(文字列,先頭位置,長さ)

□ 引数

変数 : 文字列を参照している変数または配列変数

先頭位置 : 切り出す文字位置先頭 1～32767

長さ : 切り出す文字の長さ 1～32767

文字列 : ダブルクォーテーションで囲った文字列定数(“文字列”)

□ 戻り値

指定した文字列および変数が参照している文字列を全部または一部を切り出して返します。

□ 説明

指定した文字列および変数が参照している文字列を全部または一部を切り出して返します。

引数に先頭位置、長さを指定した場合は、その条件にて文字列を切り出して返します。

先頭位置は 1 からの指定となります。

本関数は PRINT、MSG、GPRINT の引数にて利用可能です。

例:

```
10 S="123456789"
20 ?STR$(S)
30 ?STR$(S,5,2)
40 ?STR$("ABCDE",5,1)
```

実行結果

```
run
123456789
56
E
OK
```

上記の例では、変数 S が参照している文字列“123456789”に対して、
 20 行は全て出力、30 行は 5 番目からの文字から 2 文字 “56”を出力、
 40 行は直接指定した文字列“ABCDE”の 5 番目の文字から 1 文字を出力しています。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Illegal value : 指定した文字位置、長さの値が不当
 '(' or ')' expected : '(' または ')'が無い
 Overflow : 指定した数値が-32768 ～ 32767 を超えている

□ 利用例

変数が参照している文字列を切り出して表示する

```
10 'モリツリサ サンプル
20 S="Hello,Tiny BASIC"
30 L=LEN(S)
40 PRINT STR$(S);" LEN=";L
50 PRINT STR$(S,1,5)
60 C=ASC(S,12)
RUN
Hello,Tiny BASIC LEN=16
Hello
```

6.23 WSTR\$ 変数が参照する全角を含む文字列の取得・文字列の切り出し

□ 書式

WSTR\$(変数)

WSTR\$(変数,先頭位置,長さ)

WSTR\$(文字列)

WSTR\$(文字列,先頭位置,長さ)

□ 引数

変数 : 文字列を参照している変数または配列変数

先頭位置 : 切り出す文字位置先頭 1~32767

長さ : 切り出す文字の長さ 1~32767

文字列 : ダブルクォーテーションで囲った文字列定数(“文字列”)

□ 戻り値

指定した文字列および変数が参照している文字列を全部または一部を切り出して返します。

STR\$()と異なり、全角 1 文字は 1 文字としてカウントします。

□ 説明

指定した文字列および変数が参照している文字列を全部または一部を切り出して返します。

引数に先頭位置、長さを指定した場合は、その条件にて文字列を切り出して返します。

先頭位置は 1 からの指定となります。

本関数は PRINT、MSG、GPRINT の引数にて利用可能です。

例:

```
10 S="Hello,world! こんにちは世界!"
20 ?WSTR$(S)
30 ?WSTR$(S,14,8)
40 ?WSTR$("あいうえお",5,1)
```

実行結果

```
run
Hello,world! こんにちは世界!
こんにちは世界!
お
OK
```

上記の例では、変数 S が参照している文字列“Hello,world! こんにちは世界!”に対して、
 20 行は全て出力、30 行は 14 番目からの文字から 8 文字 “こんにちは世界!”を出力、
 40 行は直接指定した文字列“あいうえお”の 5 番目の文字から 1 文字 “お”を出力しています。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 指定した文字位置、長さの値が不当

'(' or ')' expected : '(' または ')'が無い

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

変数が参照している文字列を切り出して表示します。半角文字の場合は LEN() と同じ動作となります。

```
10 'tiny BASIC サンプル
20 S="Hello,Tiny BASIC"
30 L=WLEN(S)
40 PRINT STR$(S); " LEN=";L
50 PRINT STR$(S,1,5)
60 C=ASC(S,12)
RUN
Hello,Tiny BASIC LEN=16
Hello
```

6.24 RND 乱数の発生（数値関数）

□ 書式

RND(値)

□ 引数

値 : 0 ~ 32767

式、変数、配列変数、数値、数値定数の指定が可能

□ 戻り値

0 から指定した値未満の乱数

□ 説明

0 から指定した値-1 の範囲の乱数を発生させ、その値を返します。

R=RND(10)

の場合、変数 R には 0~9 範囲の値が代入されます。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Overflow : 指定した数値が-32768 ~ 32767 を超えている
 '(' or ')' expected : '(' または ')'が無い

□ 利用例

ジャンケンの表示

```

10 @(0)="グー","チョキ","ハロー"
20 FOR I=1 TO 5
30 R=RND(3)
40 PRINT I,":";STR$(@(R))
50 NEXT I
run
1:チョキ
2:ハロー
3:チョキ
4:グー
5:ハロー
OK

```

6.25 WAIT 時間待ち

□ 書式

WAIT 待ち時間(ミリ秒)

□ 引数

待ち時間： 0 ～ 32767 （単位 ミリ秒）

□ 説明

引数で指定した時間(ミリ秒単位)、時間待ち(ウェイト)を行います。

最大で 32767 ミリ秒（32.8 秒）の時間待ちが可能です。

長い時間待ちを行う必要がある場合は、TICK()や GETTIME を使った方法を検討してください。

（注意） 時間待ち中はキー操作によるプログラム中断を行うことは出来ません。

短い時間の指定にてループ処理を行う等の対策を行ってください。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 待ち時間に範囲外の値を指定した
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

□ 利用例

画面上の指定位置に時刻を 1 秒間隔で更新表示する

```

10 CLS
20 C=0
30 LOCATE 0,0
40 PRINT #-5,C
50 WAIT 1000
60 C=C+1
70 GOTO 30

```


6.26 TICK 経過時間取得（数値関数）

経過時間を取得する。

□ 書式

TICK()

TICK(モード)

□ 引数

モード 0：経過時間をミリ秒単位で取得する（デフォルト）

1：経過時間を秒単位で取得する

□ 戻り値

モード指定に従った経過時間を返す。

モード 0： 0 ～ 32767 ミリ秒の経過時間を返す(約 32.767 秒でオーバーフロー)

モード 1： 0 ～ 32767 秒の経過時間を返す(約 9.1 時間でオーバーフロー)

□ 説明

起動からの経過時間を返します。経過時間は CLT コマンドにて初期化が可能です。

CLT コマンドと組み合わせて使うことで、処理時間の測定を行うことができます。

モード指定無し、または 0 を指定の場合、ミリ秒単位の経過時間を返します。

1 を指定した場合は、秒単位の経過時間を返します。

（注意）オーバーフローが発生しますので、長時間の測定には利用することが出来ません。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
'(' or ')' expected	: '(' または ')' がない、引数の指定が正しくない
Illegal value	: モードの指定値が正しくない
Overflow	: モードの指定値が -32767 ～ 32767 を超えている

□ 利用例

1 万回ループの実行時間を測定する（プログラム：TICK.BAS）

```
10 A=0
20 T=TICK()
30 FOR I=1 TO 10000
40 A=A+1
50 NEXT I
60 PRINT A
70 PRINT TICK()-T;"msec"

RUN
10000
785msec
```

6.27 POKE 指定アドレスへのデータ書き込み

□ 書式

POKE 仮想アドレス,データ

POKE 仮想アドレス,データ,データ, ... データ (可変個数指定)

□ 引数

仮想アドレス : 仮想アドレス(16ビット) \$0000 ~

データ : 書き込むデータ(下位8ビットのみ有効)

□ 説明

指定した仮想アドレスに指定したデータを書き込みます。

仮想アドレスの指定には次の定数を利用することで有用な領域への書き込みが簡単に行えます。

VRAM	: 画面表示用メモリ
VAR	: 変数領域
ARRAY	: 配列変数領域(@ (0) ~ @ (15))
PRG	: プログラム領域
MEM	: ユーザーワーク領域

仮想アドレスの詳細については、「1.7 メモリーマップ」の「仮想アドレス」を参照下さい。

ユーザーワーク領域は利用者が自由に利用出来る領域です。

それ以外の領域は BASIC の実行にて利用する領域です。

指定した仮想アドレスに上記以外の領域を指定した場合はエラーとなります。

例として変数 X 値格納アドレスに 2 バイトデータを書き込んでみます。

```

10 X=0
20 POKE VAR+(ASC("X")-ASC("A"))*2,$34,$12
30 PRINT HEX$(X,2)
run
1234
OK

```

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Out of range value : 領域外のアドレスを指定した

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

変数 A から Z の内容をユーザーワーク領域に保存する

```

10 FOR I=0 TO 51
20 POKE MEM+I,PEEK(VAR+I)
30 NEXT I

```

6.28 PEEK 指定アドレスの値参照（数値関数）

□ 書式

PEEK(仮想アドレス)

□ 引数

仮想アドレス： 参照を行う仮想アドレス(16ビット) \$0000 ～

□ 戻り値

指定した仮想アドレスに格納されている 1 バイトデータ（0～255）

□ 説明

指定した仮想アドレスに格納されている値（1バイト）を返します。

仮想アドレスの指定には次の定数を利用することで有用な領域への参照が簡単に行えます。

VRAM	: 画面表示用メモリ
VAR	: 変数領域
ARRAY	: 配列変数領域(@ (0) ～ @ (15))
PRG	: プログラム領域
MEM	: ユーザーワーク領域

仮想アドレスの詳細については、「1.7 メモリーマップ」の「仮想アドレス」を参照下さい。

ユーザーワーク領域は利用者が自由に利用出来る領域です。

それ以外の領域は BASIC の実行にて利用する領域です。

例として画面上のカーソル位置 X,Y に格納されている文字の参照は次のようになります。

C = PEEK(VRAM+Y*48+X)

エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Out of range value	: 領域外のアドレスを指定した
'(' or ')' expected	: 括弧の指定が正しくない
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

□ 利用例

変数 A から Z の内容をユーザーワーク領域に保存する

```
10 FOR I=0 TO 51
20 POKE MEM+I, PEEK(VAR+I)
30 NEXT I
```

6.29 PRINT 画面への文字表示

□ 書式

PRINT

PRINT 文字列|数値

PRINT 文字列|数値;

PRINT 文字列|数値; 文字列|数値;...

PRINT 文字列|数値; 文字列|数値;... ;

PRINT #桁数,文字列|数値; 文字列|数値;... ;

※ ...は可変指定を示す

※ ; は連結指定、カンマ','も可能

□ 引数

文字列 : 文字列定数または文字列関数

数値 : 数値定数、変数、配列変数、または数値関数、式

連結指定 : セミicolon';' または カンマ','
文末に付けると改行抑制される桁数 : #数値 または #-数値 の形式で指定する
例:#3 、 #-3

□ 説明

指定した文字列、式、関数、変数、数値をスクリーン画面のカーソル位置に表示します。

```
PRINT "Hello,World"
Hello,World
OK
PRINT 123*3
369
OK
PRINT HEX$(123*3)
171
OK
```

文字列、式、関数、変数、数値は連結指定のセミicolon';'、またはカンマ','にて連結して表示することが出来ます。
また最後に';'または','が付加されている場合は改行しません。

```
10 N=5:C=3:K=10
20 PRINT "N=";N;" C=";C;
30 PRINT " K=";
40 PRINT K
RUN
N=5 C=3 K=10
OK
```

数値の整形表示

#数値にて桁数を指定することで、任意の桁数（指定桁に満たない場合は空白を入れる）にて等間隔で表示します。

数値の前に-（マイナス）を付加した場合、空白をではなく、0(零)で不足桁を補います。

#数値は任意の位置、任意の回数指定できます。

桁数指定なし

```
PRINT 1;" ":"2;" ":"3
1:2:3
```

桁数指定あり(空白文字で補間)

```
PRINT #2,1;" ":"2;" ":"3
1: 2: 3
```

桁数指定あり(0で補間)

```
PRINT #-2,1;" ":"2;" ":"3
01:02:03
```

表示位置の指定

LOCATE コマンドを併用することで、スクリーン画面の任意の位置に文字を表示することが出来ます。

```
10 LOCATE 10,2
20 PRINT "Hello!"
```

色、属性の指定

スクリーン画面が SCREEN 0 のターミナルコンソールの場合、

COLOR コマンド、ATTR コマンドを併用することで、文字の前景色、背景色の指定、点滅、アンダーライン等の属性を付加することが出来ます。

```
10 COLOR 4,3
20 PRINT "Hello,";
40 ATTR 2
50 PRINT "World".
60 COLOR 7,0:ATTR 0
```

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した
 Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

時刻形式で表示する

```
10 A=10:B=52:C=00
20 LOCATE 0,0:PRINT #-2,A;":";B;":";C
```

実行結果

```
10:52:00
```

6.30 INPUT 数値の入力

□ 書式

INPUT 変数

INPUT 変数,オーバーフロー時の既定値

INPUT プロンプト, 変数

INPUT プロンプト, 変数, オーバーフロー時の既定値

□ 引数

変数 : 入力した値を格納する変数または配列変数

プロンプト : 文字列定数 “プロンプト”

オーバーフロー時の既定値 : -32768 ~ 32767

□ 説明

現在のカーソル位置にて数値の入力をし、指定した変数にその値を格納します。

キーボードから入力できる文字は符号-、+、数値 0~9、入力訂正の[BS]、[DEL]、入力確定の[Enter]です。

それ以外の入力はできません。

~~数値を入力せずに、[Enter]にて入力確定を行った場合、変数には0が格納されます。~~

引数に変数のみを指定した場合、“変数名:”を表示しその後ろの位置から数値を入力します。

```
INPUT A
A:
```

プロンプトを指定した場合は、そのプロンプトを表示しその後ろ位置から数値を入力します。

```
INPUT "Value=",A
Value=
```

オーバーフロー時の既定値を指定した場合、入力値した数値がオーバーフローを発生した場合は、オーバーフロー時の既定値を変数に設定します。オーバーフロー時の既定値を設定していない場合は、Overflow エラーとなります。

オーバーフロー時の既定値なし

```
INPUT A
A:111111
Overflow
OK
```

オーバーフロー時の既定値あり

```
INPUT A, -1
A:111111
OK
?A
-1
OK
```

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

```
INPUT A
A=1234
OK

INPUT "Value=" A
Value=1234
OK
```

6.31 INKEY キー入力の読み取り（数値関数）

□ 書式

INKEY()

□ 引数

なし

□ 戻り値

押したキーの文字コード 0 ～ 255

キーが押されていない場合は 0 を返します

□ 説明

キーボード上の押しているキーのコードを取得します。キーが押されていない場合は、0 を返します。

押したキーが返すコードについては「エラー! 参照元が見つかりません。エラー! 参照元が見つかりません。」を参照下さい。

（注意） [ESC]、[CTRL-C]はプログラム中断用のため、キー入力の読み取りは出来ません。
 キーボードにおいて利用出来ないキーはコードの取得が出来ません。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

□ 利用例

入力したキーのコードを調べる

```
10 CLS
20 I= INKEY()
30 IF I>0 LOCATE 0,0:?"#3,I
40 GOTO 20
```

6.32 CLS 画面表示内容の全消去

□ 書式

CLS

CLS 消去対象画面

□ 引数

なし

□ 説明

画面上に表示している内容を全て消します。

COLOR コマンドで色指定した場合、文字色、背景色の設定を全画面に反映されます。

□ エラーメッセージ

Illegal value : 指定した引数が有効範囲でない

Overflow : 指定した数値が-32768 ~ 32767 を超えている

Syntax error : 文法エラー、書式と異なる利用を行った

□ 利用例

画面の表示内容を消去する

CLS
OK

6.33 COLOR 文字色の設定

□ 書式

COLOR 文字色

COLOR 文字色, 背景色

□ 引数

文字色: 色コード 0～8

背景色: 色コード 0～8

□ 説明

シリアルコンソールの文字色の設定を行います。指定した色は以降の文字表示に反映されます。

文字色、背景色で指定する色コードに対する色は次の表の通りです。

表 1 色コード

色コード	色
0	黒
1	赤
2	緑
3	茶
4	青
5	マゼンタ
6	シアン
7	白(デフォルト)
8	黄

(注意) 利用するターミナルソフトにより色が正しく表示されない場合があります。

属性指定との併用では正しく表示されない場合があります。

画面を[F5]、[CTRL-R]、[Page UP]、[Page Down]、[BS]、[DEL]キー等にて文字編集による再表示が称した場合、色情報は欠落します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Illegal value : 色コードに範囲外の値を指定した
 Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

ランダムな色で*を表示する

```
10 FOR I=0 TO 10
20 FOR J=0 TO 10
30 COLOR RND(8): ? "*";
35 WAIT 100
40 NEXT J
50 ?
60 NEXT I
```

6.34 ATTR 文字表示属性の設定

□ 書式

ATTR 属性

□ 引数

属性：属性コード 0 ～ 4

□ 説明

シリアルコンソールの文字の表示属性を設定します。指定した表示属性は以降の文字表示に反映されます。

属性に指定する属性コードは次の表の通りです。

表 2 属性コード

属性コード	機能
0	標準(デフォルト)
1	下線
2	反転
3	ブリンク
4	ボールド

(注意) 利用するターミナルソフトにより色が正しく表示されない場合があります。
属性指定との併用では正しく表示されない場合があります。
画面を[CTRL-R]、[Page UP]、[Page Down]キーにて再表示した場合、
色情報は欠落します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
Illegal value : 属性コードに範囲外の値を指定した
Overflow : 指定した数値が-32768 ～ 32767 を超えている

□ 利用例

Hello,world を反転表示する

```
10 CLS
20 LOCATE 5,5
30 ATTR 2:?"Hello,world"
40 ATTR 0
```

6.35 LOCATE カーソルの移動

□ 書式

LOCATE 横位置, 縦位置

□ 引数

横位置: 画面上の横位置 0 ~ 47

縦位置: 画面上の縦位置 0 ~ 7

□ 説明

カーソルを指定した位置に移動します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

画面の指定位置にメッセージを表示する

```
10 CLS
20 LOCATE 5,0:PRINT "Hello,world."
30 LOCATE 6,1:PRINT "TinyBASIC"
40 LOCATE 7,2:PRINT "Thank you."
```

実行結果

```
Hello,world.
 TinyBASIC
  Thank you.
```

6.36 VCLS VFD 表示内容の全消去

□ 書式

VCLS

□ 引数

なし

□ 説明

VFD の表示内容を消去します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

VFD に「こんにちは、世界！」の表示と消去を繰り返します。

```

10 VMSG 0,"こんにちは、世界！"
20 WAIT 500
30 VCLS
40 WAIT 500
50 GOTO 10

```

□ 補足

VFD の表示内容を一時的に見えなくする方法として次の別の方法があります。

- ① 輝度を 0 にする

VBRIGHT 0

- ② 表示を OFF にする

VDISPLAY OFF

上記の場合、VFD の表示内容自体は保持されています。

①の場合は、VBRIGHT 255、②の場合は VDISPLAY ON で表示を ON にすれば再び表示出来ます。

6.37 VDISPLAY VFD 表示のオン・オフ

□ 書式

VDISPLAY モード

□ 引数

モード : ON 点灯、OFF 消灯 または 1 点灯 0 消灯

□ 説明

VFD の表示のオン・オフ設定を行います。

VBRIGHT で輝度設定を行っている場合は、表示をオン時には直前の輝度にて表示されます。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

VFD に「こんにちは、世界！」の表示と消去を繰り返します。

```

10 VCLS
20 VMSG 0,"こんにちは、世界！"
30 WAIT 500
40 VDISPLAY OFF
50 WAIT 500
60 VDISPLAY ON
70 GOTO 30

```

6.38 VBRIGHT VFD 輝度設定

□ 書式

VBRIGHT 輝度

□ 引数

輝度 : 0(0%) ~ 255(100%)

□ 説明

VFD の表示の輝度を設定します。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

VFD に「こんにちは、世界！」の表示をフェーフォアウトします。

```

10 VCLS
20 VMSG 0,"こんにちは、世界！"
30 VBRIGHT 10
40 FOR I=100 TO 0 STEP -10
50 VBRIGHT I
60 WAIT 50
70 NEXT I
80 WAIT 500
90 VBRIGHT 255
100 GOTO 40

```

6.39 VMSG VFD にメッセージ文の表示

□ 書式

VMSG 速度, メッセージ文

□ 引数

速度 : メッセージ文のスクロール速度 0 ~ 1024(ミリ秒)、または-1
 0の場合は、即表示、-1の場合はデータ送信のみ(待ち時間なし、ラッチなし)
 メッセージ文 : 文字列、変数、数値、関数を結合した文(PRINT コマンドのフォーマット)

□ 説明

VFD に指定したメッセージ文を表示します。メッセージ文には全角文字(シフト JIS)の利用が可能です。

メッセージ文は PRINT コマンドと同じフォーマットによる指定が可能です。

速度には左スクロールする 1 ドット単位のウェイト時間(ミリ秒)を指定します。

0を指定した場合、即時表示します。

-1 を指定した場合、データの送信のみで表示には反映されません。次の表示(ラッチ)のタイミングで表示に反映されます。

速度に-1 を指定してデータ送信のみを行った後、速度に 0 以上を指定することで一緒に表示に反映されます。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Illegal value : 範囲外の値を指定した
 Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

配列に格納した文字列を切り替えて表示する

```
10 'VDF表示サンプル
20 TONE 440,100:TONE 880,100
30 @(0)="熱いぞ! 埼玉!"
40 @(1)="豊四季タイニーBASIC"
50 @(2)="VFD対応バージョン"
60 @(3)="只今開発中!"
70 @(4)="ちょっとメモリー的に厳しい.."
80 VBRIGHT 255
90 FOR S=0 TO 4
100 VCLS:VMSG 10,STR$(@(S))
110 FOR I=0 TO 255 STEP 15
120 VBRIGHT 255-I:WAIT 40
130 NEXT I
140 WAIT 500
150 VBRIGHT 255:WAIT 500
160 VSCROLL 256,3
170 WAIT 500
180 NEXT S
190 GOTO 80
```

□ 補足

データ送信のみを利用することで、左から右にスクロールする表示を行うことが出来ます。

```
10 VCLS
20 S="こんにちは、世界!"
30 L=LEN(S)
40 FOR I=256 TO 1 STEP -2
50 VMSG -1,STR$(S)
60 VSCROLL I,0
70 VSCROLL 256,-1
80 NEXT I
90 VBRIGHT 255
100 GOTO 40
```

6.40 VSCROLL VFD に指定長空白の表示

□ 書式

VSCROLL スクロール幅, 速度

□ 引数

スクロール幅 : 1~256 (ドット)

速度 : スクロール速度 0 ~ 1024(ミリ秒)、または-1

0 の場合は、即表示、-1 の場合はデータ送信のみ（待ち時間なし、ラッチなし）

□ 説明

VFD に指定長の空白をスクロール表示することで、表示内容を左にスクロールさせます。

0 を指定した場合、即時表示します。

-1 を指定した場合、データの送信のみで表示には反映されません。次の表示（ラッチ）のタイミングで表示に反映されます。

速度に-1 を指定してデータ送信のみを行った後、速度に 0 以上を指定することで一緒に表示に反映されます。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 範囲外の値を指定した

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

配列に格納した文字列を切り替えて表示する

```

10 'VDF表示サンプル
20 TONE 440,100:TONE 880,100
30 @(0)="熱いぞ！埼玉！"
40 @(1)="豊四季タイニーBASIC"
50 @(2)="VFD対応バージョン"
60 @(3)="只今開発中！"
70 @(4)="ちょっとメモリー的に厳しい.."
80 VBRIGHT 255
90 FOR S=0 TO 4
100 VCLS:VMSG 10,STR$(@S))
110 FOR I=0 TO 255 STEP 15
120 VBRIGHT 255-I:WAIT 40
130 NEXT I
140 WAIT 500
150 VBRIGHT 255:WAIT 500
160 VSCROLL 256,3
170 WAIT 500
180 NEXT S
190 GOTO 80

```

□ 補足

データ送信のみを利用することで、左から右にスクロールする表示を行うことができます。

```

10 VCLS
20 S="こんにちは、世界！"
30 L=LEN(S)
40 FOR I=256 TO 1 STEP -2
50 VMSG -1,STR$(S)
60 VSCROLL I,0
70 VSCROLL 256,-1
80 NEXT I
90 VBRIGHT 255
100 GOTO 40

```


6.41 VPUT VFD に直接データ送信

□ 書式

VPUT データ格納アドレス, データ長, 速度

□ 引数

データ格納アドレス : ビットマップデータ格納アドレス

データ長 : 1~512 (バイト)

速度 : スクロール速度 0 ~ 1024(ミリ秒)、または-1
: 0の場合は、即表示、-1の場合はデータ送信のみ(待ち時間なし、ラッチなし)

□ 説明

VFD に直接、データ(無加工データ)を送信します。

送信するデータはデータ格納アドレスに指定します。

速度に0を指定した場合、即時表示します。

-1 を指定した場合、データの送信のみで表示には反映されません。次の表示(ラッチ)のタイミングで表示に反映されます。

速度に-1 を指定してデータ送信のみを行った後、速度に0以上を指定することで一緒に表示に反映されます。

□ エラーメッセージ


Syntax error : 文法エラー、書式と異なる利用を行った
Illegal value : 範囲外の値を指定した
Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

配列@(0)の値のビット位置を変化させて、その値(2バイト)のアドレスから2バイトを表示用データとして送信します。

```
1 'VFDジグザグ表示
10 VCLS
20 @(0)=1
30 FOR I=0 TO 14
40 VPUT ARRAY,2,5
50 @(0)=@(0)<<1
60 NEXT I
70 FOR I=0 TO 14
80 @(0)=@(0)>>1
90 VPUT ARRAY,2,5
100 NEXT I
110 GOTO 30
```

実行結果



6.42 TONE 単音出力

□ 書式

TONE 周波数

TONE 周波数,出力期間

□ 引数

周波数 : 0 ~ 32767 (Hz) 0 の場合は消音

出力期間 : 0 ~ 32767 (ミリ秒) 0 の場合は、継続再生

□ 説明

指定した周波数の単音出力を行います。

デフォルトの設定では **PN8 ピン** を単音出力として使用します。

単音出力ピンに圧電スピーカー（圧電サウナダ）を接続すること音を出すことが出来ます。

SETTONE コマンドにて利用するピンの変更が可能です。

出力期間の指定がある場合は、その期間パルスを出力します(ミリ秒単位)。

出力期間の指定がある場合、出力完了待ちを行います。

出力期間の指定がない場合は、NOTONE コマンドで停止指示をするまでパルスを出力し続けます。

音階・周波数対応表

	ド	ド#	レ	レ#	ミ	ファ	ファ#	ソ	ソ#	ラ	ラ#	シ
1	33	35	37	39	41	44	46	49	52	55	58	62
2	65	69	73	78	82	87	93	98	104	110	117	123
3	131	139	147	156	165	175	185	196	208	220	233	247
4	262	277	294	311	330	349	370	392	415	440	466	494
5	523	554	587	622	659	698	740	784	831	880	932	988
6	1047	1109	1175	1245	1319	1397	1480	1568	1661	1760	1865	1976
7	2093	2217	2349	2489	2637	2794	2960	3136	3322	3520	3729	3951
8	4186	4435	4699	4978	5274	5588	5920	6272	6643	7040	7459	7902

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

スペースキーを押したら音を鳴らす

```

10 IF INKEY() = 32 TONE 880,50
20 GOTO 10

```

6.43 NOTONE 単音出力停止

□ 書式

NOTONE

□ 引数

なし

□ 説明

TONE コマンドによるパルス出力を停止します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

□ 利用例

音を停止する

```
10 TONE 400  
20 WAIT 200  
30 NOTONE
```

6.44 PLAY 音楽演奏(MML 文)

□ 書式

PLAY “MML 文”

□ 引数

MML 文 : MML を定義した文字列

□ 説明

※本コマンドは、コンパイルオプションの設定にて利用可能となります（デフォルトでは利用不可）

MML (Music Macro Language)にて定義した文法を元に、接続している圧電スピーカーを使って単音による、音楽演奏を行います。音の出力はデフォルトでは**ピン番号 8**を利用します。SETTONE コマンドにて変更が可能です。

MML 文には次のコマンドを利用することが出来ます。

- 音階記号 [#|+|-][長さ][.]
 音階指定 : C、D、E、F、G、A、B ~ または c、d、e、f、g、a、b
 順番に ド、レ、ミ、ファ、ソ、ラ、シ、ド の音階に対応します。
 # : 半音上げる (省略可能)
 + : 半音上げる (省略可能)
 - : 半音下げる (省略可能)
 長さ : 1、2、4、8、16、32、64 (省略時は L による長さ、デフォルト値 4)、省略可能
 1 は全音符、2 は 2 分音符、4 は四分音符、64 は 64 分の一音符
 . : 長さを半分伸ばす
- R[長さ] : 休符
 長さ : 1、2、4、8、16、32、64 (省略時は L による長さ、デフォルト値 4)、省略可能
 1 は全音符、2 は 2 分音符、4 は四分音符、64 は 64 分の一音符
 . : 長さを半分伸ばす
- L<長さ>
 音の長さを指定します。省略時の長さの指定。初期値は 4 (四分音符)
 長さ : 1、2、4、8、16、32、64 (省略時は L による長さ、デフォルト値 4)、省略可能
 1 は全音符、2 は 2 分音符、4 は四分音符、64 は 64 分の一音符
 . : 長さを半分伸ばす
- O<オクターブ>
 音の高さを指定します。
 オクターブ : 1~8 初期値は 4
- < : 1 オクターブ上げる
- > : 1 オクターブ下げる
- T<テンポ> : テンポを指定する。初期値は 120、32~255
- 空白文字 : スキップします。

演奏の中断は[ESC]キー 2 回押し、または[CTRL]+C キーにて行うことが出来ます。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Illegal MML : MML 文の文法エラー
 Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

「ドラミファソラシド」と演奏する。

```
PLAY "CDEFGAB<C"
OK
```

「ねこふんじゃった」を演奏する。

※MML は、下記のサイトに公開されているものを利用しています。

- 主体性の無いページ (<http://astr.me.land.to/>)

MML (<http://astr.me.land.to/tool/mabi/>)

ねこふんじゃった! (<http://astr.me.land.to/tool/mabi/mml/nekof.htm>)

```
10 'ネコフツ' ャッタ
20 'リョウモト http://astr.me.land.to/tool/mabi/mml/nekof.htm
60 PLAY "L16D+C+R8F+RF+RD+C+R8F+RF+RD+C+L8RF+RF+RL16FRFRD+C+R8FRFRD+C+R8FRFRD+C+"
70 PLAY "L8RFRFRL16F+RF+RD+C+R8F+RF+RD+C+R8F+RF+RD+C+L8RF+RF+RL16FRFRD+C+R8FRFRD+C+R8FR"
80 PLAY "L16FRD+C+L8RFRFRL16F+RF+RD+C+L8RF+RF+RF+RF+RF+RF+RL16FRFRD+C+L8RFRFRFRFRFR"
90 PLAY "L16F+RF+RD+C+R8F+RF+RD+C+R8F+RF+RD+C+L8RF+RF+RL16FRFRD+C+R8FRFRD+C+R8FRFRD+C+"
100 PLAY "L8RFRFRL16F+RF+R8.F+RC+C+D8C+8.FRF+"

```

6.45 GPIO GPIO 機能設定

□ 書式

GPIO ピン番号, モード

□ 引数

ピン番号 : 0 ~ 21

モード :

OUTPUT : デジタル出力

INPUT_FL : デジタル入力 (フLOAT状態 : Arduino の INPUT 指定と同じ)

INPUT_PU : デジタル入力 (内部プルアップ抵抗有効)

□ 説明

ボード上の指定したピン番号の入出力機能の設定を行います。

Arduino の pinMode() に相当します。

ピン番号の指定には、0~21 が可能です。

(注意) デフォルトでは、INPUT_FL が設定されています。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Cannot use GPIO fuinction	: ピン番号に利用出来ないモード設定を行った
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
Illegal value	: 指定した数値が不当である

□ 利用例

PNO ピンからデジタル入力値を読み取り、その値を画面に随時表示します。

```

10 CLS
20 GPIO 2, INPUT_PU
30 A=IN(2)
40 LOCATE 0,0: ? A;
50 GOTO 30

```

6.46 OUT デジタル出力

□ 書式

OUT ピン番号, 出力値

□ 引数

ピン番号 : 0 ~ 21

出力値 :

LOW または 0 : 0V を出力する

HIGH or 0 以外の値 : 5V を出力する

□ 説明

指定ピンから、指定した出力を行います。

出力を行う場合は事前に GPIO コマンドによる機能設定（出力設定）が必要です。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Cannot use GPIO fuinction	: ピン番号に利用出来ないモード設定を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

ピン番号 2 に接続している LED を点滅させます。

```

10 'L 灯
20 GPIO 2,OUTPUT
30 OUT 2,HIGH
40 WAIT 300
50 OUT 2,LOW
60 WAIT 300
70 GOTO 30

```

6.47 POUT PWM パルス出力

□ 書式

POUT ピン番号, デューティ値

□ 引数

ピン番号 : 3, 5, 6, 9, 10, 11

デューティ値 : 0 ~ 255

0 がデューティ比 100%となります。

255 がデューティ比 100%となります。

□ 説明

指定ピンから、PWM パルス出力を行います。周波数は 490Hz となります(Arduino Uno 互換) (要検証)

パルス出力を停止する場合は、デューティ値に 0 を指定して下さい。

□ エラーメッセージ


Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

ピン A6 (No.20) に接続したスライド抵抗器でピン No.6 に接続している LED の明るさを制御します。

```
10 A=ANA(20)>>2
20 IF A<10 A=0
30 POUT 6,A
40 GOTO 10
```

実行結果



6.48 IN デジタル入力（数値関数）

□ 書式

IN(ピン番号)

□ 引数

ピン番号 : 0 ~ 21

□ 戻り値

取得した値 0 (LOW) または 1 (HIGH)

□ 説明

指定ピンの入力値を読み取り、その値を返します。

入力を行う場合は事前に GPIO コマンドによる機能設定（入力設定）が必要です。

□ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Cannot use GPIO fuinction	: ピン番号に利用出来ないモード設定を行った
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
'(' or ')' expected	: '(' または ')'がない

□ 利用例

ピン番号 2 からデジタル入力値を読み取り、その値を画面に随時表示します。

```

10 CLS
20 GPIO 2, INPUT_PU
30 A=IN(2)
40 LOCATE 0,0: ? A;
50 GOTO 30

```

6.49 ANA アナログ入力（数値関数）

□ 書式

ANA(ピン番号)

□ 引数

ピン番号： 14,15,16,17,18,19,20,21 （Arduino の A0～A7 に対応）

□ 戻り値

取得した値 0～1023(10 ビット)

□ 説明

指定ピンのアナログ入力値を読み取り、その値を返します。

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Illegal value : ピン番号、モードに範囲外の値を指定した
 Overflow : 指定した数値が-32768 ～ 32767 を超えている
 '(' or ')' expected : '(' または ')'がない


□ 利用例

□ 利用例

ピン A6（No.20）に接続したスライド抵抗器で VFD に表示している縦線の表示位置を操作します。

```
10 @(0)=$FFFF
20 A=ANA(20)>>2:IF A=0 A=1
30 VPUT ARRAY,2,-1
40 VSCROLL A,0
50 VSCROLL 256,-1
60 WAIT 20
70 GOTO 20
```

実行結果



VFD に表示されている縦線がスライド抵抗の位置に応じて移動します。

6.50 I2CR I2C スレーブデバイスからのデータ受信（数値関数）

□ 書式

I2CR(デバイスアドレス,コマンドアドレス,コマンド長,受信データアドレス,データ長)

□ 引数

デバイスアドレス : 0 ~ 127 (\$00 ~ \$7F) (7ビット指定)
 I2C スレーブアドレスを7ビット形式で指定
 コマンドアドレス : 0 ~ 32767 (\$0000 ~ \$1FFFF)
 送信するコマンドが格納されている仮想アドレス
 コマンド長 : 0 ~ 32767
 送信するコマンドのバイト数
 受信データアドレス : 0 ~ 32767 (\$0000 ~ \$1FFFF)
 受信データを格納するSRAM内仮想アドレス
 データ長 : 0 ~ 32767
 受信するデータのバイト数

□ 説明

I2C スレーブデバイスからデータを受信します。

送信先はデバイスアドレスにてI2C スレーブアドレスを指定します。

受信において、コマンド等の制御データの送信が必要な場合は、コマンドアドレス、コマンド長にて送信するデータを指定します。受信のみを行う場合は、コマンドアドレス、コマンド長に0を設定します。送信受信するデータは仮想アドレスにて指定します。

仮想アドレスの指定には次の定数を利用することで有用な領域への参照が簡単に行えます。

VRAM : 画面表示用メモリ
 VAR : 変数領域
 ARRAY : 配列変数領域
 PRG : プログラム領域
 MEM : ユーザーワーク領域

I2C 通信には次の接続ピンを利用します。

18 : (Arduino A4 ピン SDA)
 19 : (Arduino A5 ピン SDA)

□ 戻り値

0 : 正常終了
 1 : 通信バッファに対してデータが長すぎる
 2 : アドレス送信にNACKが返された
 3 : データ送信にNACKが返された
 4 : その他のエラー

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
 Illegal value : 指定した引数の値が不当である
 Out of range value : 指定した値が有効範囲を超えている
 Overflow : 指定した数値が-32768 ~ 32767を超えている

□ 利用例

I2C EEPROM(AT24C256 スレーブアドレス \$50)にデータの読み書きを行う

```
100 POKE MEM+0,$00,$00
110 POKE MEM+2,64,65,66,67
120 R=I2CW($50,MEM,2,MEM+2,4)
130 ? "snd r=";R
140 POKE MEM+6,0,0,0,0
150 WAIT 5
160 R=I2CR($50,MEM,2,MEM+6,4)
170 ? "rcv r=";R
180 ? "rcv data:";
190 FOR I=0 TO 3
200 ? PEEK(MEM+6+I);" ";
210 NEXT I
220 ?
```

実行結果

```
run
snd r=0
rcv r=0
rcv data:64 65 66 67
OK
```

6.51 I2CW I2C スレーブデバイスへのデータ送信（数値関数）

□ 書式

I2CW(デバイスアドレス,コマンドアドレス,コマンド長,データアドレス,データ長)

□ 引数

デバイスアドレス : 0 ~ 127 (\$00 ~ \$7F) (7ビット指定)
I2C スレーブアドレスを7ビット形式で指定
コマンドアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)
送信するコマンドが格納されている SRAM 内仮想アドレス
コマンド長 : 0 ~ 32767
送信するコマンドのバイト数
データアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)
送信するデータが格納されている SRAM 内仮想アドレス
データ長 : 0 ~ 32767
送信するデータのバイト数

□ 説明

I2C スレーブデバイスにデータを送信します。

送信先はデバイスアドレスにて I2C スレーブアドレスを指定します。

送信するデータは SRAM 先頭からの仮想アドレスにて指定します。

送信するデータはあらかじめ、設定しておく必要があります。

コマンドとデータの区別はありません。デバイスに対しては、単純にコマンド、データの順に送信しています。

仮想アドレスの指定には次の定数を利用することで有用な領域への参照が簡単に行えます。

VRAM : 画面表示用メモリ
VAR : 変数領域
ARRAY : 配列変数領域
PRG : プログラム領域
MEM : ユーザーワーク領域

I2C 通信には次の接続ピンを利用します。

18 : (Arduino A4 ピン SDA)
19 : (Arduino A5 ピン SDA)

□ 戻り値

0 : 正常終了
1 : 通信バッファに対してデータが長すぎる
2 : アドレス送信に NACK が返された
3 : データ送信に NACK が返された
4 : その他のエラー

□ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った
Illegal value : 指定した引数の値が不当である
Out of range value : 指定した値が有効範囲を超えている
Overflow : 指定した数値が-32768 ~ 32767 を超えている

□ 利用例

I2C EEPROM(AT24C256 スレーブアドレス \$50)にデータの読み書きを行う

```

100 POKE MEM+0,$00,$00
110 POKE MEM+2,64,65,66,67
120 R=I2CW($50,MEM,2,MEM+2,4)
130 ? "snd r=";R
140 POKE MEM+6,0,0,0,0
150 WAIT 5
160 R=I2CR($50,MEM,2,MEM+6,4)
170 ? "rcv r=";R
180 ? "rcv data:";
190 FOR I=0 TO 3
200 ? PEEK(MEM+6+I);" ";
210 NEXT I
220 ?

```

実行結果

```

run
snd r=0
rcv r=0
rcv data:64 65 66 67
OK

```

接続しているI2C スレーブを調べる

(補足) I2C スレーブアドレスのみ送信し、正常終了(ACK を返した)のアドレスを調べています。

```

10 FOR I=0 TO $7F
20 C=I2CW(I,MEM,0,MEM,0)
30 IF C=0 PRINT HEX$(I,2);" ";
40 NEXT I
50 PRINT :PRINT "done."

```


実行結果

```

run
57 68
done.
OK

```

実行結果は、接続しRTC モジュール (I2C EEPROM 搭載) のI2C アドレスを検出しました。




7. 応用プログラム

7.1 RTC モジュール（DS3231 搭載）を使った時刻の表示

I2C 接続にて利用出来る RTC モジュールを使った時刻表示プログラムです。
プログラムは時刻設定プログラムと時刻表示プログラムに分かれています。

利用した RTC モジュール（RTC コントローラ DS3231 搭載）



RTC モジュールのと SCLMW25616L 実験用表示モジュールの結線

RTC モジュール側 SCLMW25616L 実験用表示モジュール側

GND	GND
VCC	VCC (5V)
SDA	A4
SCL	A5

RTC モジュールの時刻設定

次のプログラムをコピーしてコンソール上に張り付けて転送して下さい。

転送時の設定は、「2.7 パソコンからのプログラム読込」を参照して下さい。

RTC 時刻設定プログラム

```

10 INPUT "西暦年(下2桁)=" ,Y
20 INPUT "月(1~12)=" ,M
30 INPUT "日(1~31)=" ,D
40 INPUT "曜日(0:日~6:土)=" ,W
50 INPUT "時(0~23)=" ,H
60 INPUT "分(0~59)=" ,N
70 A=$68
80 POKE MEM,0,0,((N/10)<<4)+N %10
90 POKE MEM+3,((H/10)<<4)+H %10
100 POKE MEM+4,((D/10)<<4)+D %10
110 POKE MEM+5,((W/10)<<4)+W %10
120 POKE MEM+6,((M/10)<<4)+M %10
130 POKE MEM+7,((Y/10)<<4)+Y %10
140 R=I2CW($68, MEM, 1, MEM+1, 7)
150 IF R=0 ?"時刻の設定を完了しました":END
160 ?"時刻設定エラーです"

```

実行すると、コンソール画面上にて年、月、日、曜日、時、分をたずねてくるので入力して下さい。

正常に書き込んだ場合は、「時刻の設定を完了しました」と表示されます。

VFD への時刻表示

次のプログラムをコピーしてコンソール上に張り付けて転送して下さい。

転送時の設定は、「2.7 パソコンからのプログラム読込」を参照して下さい。


RTC 時刻表示プログラム

```
1'VFD 時刻表示
10 VCLS:VBRIGHT 255
20 @(0)="時","分","秒"
30 M=ARRAY+6:POKE M,0
40 R=I2CR($68,M,1,M+1,7)
50 VMSG -1,"現在の時刻は"
60 FOR I=0 TO 2
70 VMSG -1,HEX$(PEEK(M+3-I),2);STR$(@(I))
80 NEXT I
90 VMSG 0,"です"
100 WAIT 1000
110 VSCROLL 256,-1
120 GOTO 40
```

※輝度設定(VBRIGHT 255)を 100%以外に設定した場合、表示が乱れる場合があります。

この原因については、調査中です。

実行すると VFD に現在時刻がリアルタイムに表示されます。



ボード起動時に時刻表示プログラムを自動起動する場合は、次の設定を行います。

- ①プログラム番号 0 にプログラムを保存する。

```
save 0
OK
files
0:'VFD 時刻表示
1:(none)
OK
```

- ②ボードの No.7 のポートを VCC(5V) に接続する。

これにより、ボード起動時に時刻表示プログラムが実行されます。